

Euclid's Elements — Book I

Proposition 16

CGJteamLab

If one side of a triangle is produced, the exterior angle is greater than either of the interior and opposite angles.

1 Introduction

Euclid's Proposition I.16 is the exterior-angle inequality for a triangle. If a side of a triangle is extended beyond one of its vertices, the exterior angle formed at that vertex is greater than either of the two remote interior angles.

For a nondegenerate triangle ABC , extend BC through C to a point D , so that

$$B - C - D.$$

The proposition asserts

$$\angle BAC < \angle ACD \quad \text{and} \quad \angle ABC < \angle ACD.$$

Unlike Proposition I.15, this proposition is not merely a thin application of one previously packaged theorem. Its reconstruction requires a substantial piece of Hilbert order-and-congruence geometry: an interior ray must be transported between congruent angles, and the resulting angle comparison must be expressed without numerical angle measure.

The formal development therefore isolates the geometric machinery in the Hilbert interface and leaves the Euclidean proposition itself comparatively short. In particular, the final Lean file contains three theorems:

- `euclid_proposition_16_first`,
- `euclid_proposition_16_second`,
- `euclid_proposition_16`.

2 The Classical Configuration

Let ABC be a nondegenerate triangle and let the side BC be produced through C to D :

$$B - C - D.$$

The angle $\angle ACD$ is exterior to the triangle. The two interior opposite angles are $\angle BAC$ and $\angle ABC$.




Figure 1: Statement configuration for Proposition I.16: BC is produced beyond C to D .

The formal hypotheses are

$$\neg \text{Collinear}(A, B, C) \quad \text{and} \quad B - C - D.$$

The noncollinearity hypothesis makes the triangle nondegenerate, while strict betweenness records that D lies on the continuation of BC beyond C .

3 Statement

Euclid states Proposition I.16 as follows:

If one side of a triangle is produced, the exterior angle is greater than either of the two interior opposite angles.

In the Hilbert reconstruction, strict comparison of angles is represented by **HilbertAngleLess**. Thus the formal conclusion is

$$\text{HilbertAngleLess}(BAC, ACD) \quad \wedge \quad \text{HilbertAngleLess}(ABC, ACD).$$

More explicitly,

$$\neg \text{Collinear}(A, B, C), \quad B - C - D \implies \begin{cases} \angle BAC < \angle ACD, \\ \angle ABC < \angle ACD. \end{cases}$$

The relation **HilbertAngleLess** is synthetic. It does not compare numerical measures. An angle is smaller than another when a congruent copy of the first can be realized as a proper interior subangle of the second.

4 Classical Proof

Euclid proves the first inequality by a midpoint construction.

Let E be the midpoint of AC . Join BE and extend it beyond E to a point F such that

$$BE = EF.$$

Join CF .




Figure 2: Wilkins's auxiliary construction for the first inequality: E bisects AC , BE is produced to F with $BE \cong EF$, and AC is produced beyond C to G .

Since $AE = EC$, $BE = EF$, and the vertical angles $\angle AEB$ and $\angle CEF$ are equal by Proposition I.15, the triangles AEB and CEF are congruent by Proposition I.4. Hence

$$\angle BAE = \angle ECF.$$

Since E lies on AC , this gives

$$\angle BAC = \angle ACF.$$

But the ray CF lies inside the exterior angle ACD . Therefore

$$\angle BAC < \angle ACD.$$

This last sentence is exactly where the diagram carries hidden order information. Wilkins spells it out: F must lie on the same side of CD as A and on the same side of AC as D ; these two half-plane conditions characterize the interior of $\angle ACD$. Thus the inclusion of the ray CF in the exterior angle is a theorem, not a fact that may be read off the picture.

For the second opposite angle Euclid repeats the same argument after extending another side of the triangle. The equality of the corresponding vertical exterior angles then identifies the resulting exterior angle with $\angle ACD$.

Thus the classical architecture is

$$\text{midpoint construction} \rightarrow \text{I.15} \rightarrow \text{SAS (I.4)} \rightarrow \text{interior subangle} \rightarrow \text{I.16}.$$

5 Proof Architecture of the Reconstruction

The Hilbert reconstruction retains the synthetic structure of Euclid's argument but packages the difficult order-theoretic steps into reusable interface theorems.

statement: B-C-D

```

|
+----- first opposite angle -----+
|
v                                     |

```

```

Hilbert auxiliary construction
A-M-C, B-M-E + congruence data
  |
  v
AB || CE
  |
hilbert_exterior_angle_less
  |
  v
BAC < ACD
      +----- second opposite angle -----+
      |
extend AC beyond C: A-C-F
      |
apply first half to triangle BAC
      |
ABC < BCF
      |
vertical angles at C:
BCF ~= ACD
      |
transport strict angle order
      |
ABC < ACD
      |
      v
euclid_proposition_16

```

Here the distinction between the classical and formal diagrams matters. Wilkins's E, F construction explains Euclid's SAS argument and the nontrivial claim that F lies inside the exterior angle. The Lean helper uses its own auxiliary points M, E and packages that order geometry into the Hilbert interface. The second Lean inequality is then obtained by a genuinely new extension $A - C - F$ and vertical-angle transport.

For the first opposite angle the auxiliary construction produces points M and E with

$$A - M - C, \quad B - M - E,$$

and the required segment and angle congruences. From these data the interface proves that AB is parallel to CE . The theorem `hilbert_exterior_angle_less` then concludes directly that

$$\angle BAC < \angle ACD.$$

Schematically:

$$\begin{array}{c}
\neg \text{Collinear}(A, B, C) \\
\downarrow \\
\text{hilbert_exterior_angle_aux} \\
\downarrow \\
A - M - C, B - M - E, \text{ congruence data} \\
\downarrow \\
\text{hilbert_exterior_angle_aux_parallel} \\
\downarrow \\
AB \parallel CE \\
\downarrow \\
\text{hilbert_exterior_angle_less} \\
\downarrow \\
\angle BAC < \angle ACD.
\end{array}$$

The second inequality is deliberately reduced to the first. Extend AC through C to F . Apply the first part to triangle BAC to obtain

$$\angle ABC < \angle BCF.$$

The lines ACF and BCD intersect at C , so Proposition I.15 gives the vertical-angle congruence

$$\angle BCF \cong \angle ACD.$$

Transporting the right-hand side of the strict angle comparison across this congruence yields

$$\angle ABC < \angle ACD.$$

Thus the second half has the compact architecture

$$\text{first half of I.16} \rightarrow \text{I.15} / \text{vertical angles} \rightarrow \text{transport of } < \rightarrow \text{second half of I.16.}$$

6 Hilbert Reconstruction

6.1 The First Opposite Interior Angle

The theorem `euclid_proposition_16_first` begins from

$$\neg \text{Collinear}(A, B, C), \quad B - C - D.$$

It invokes

$$\text{hilbert_exterior_angle_aux}$$

to obtain points M, E and the auxiliary configuration required for the exterior-angle argument.

The same configuration is passed to

$$\text{hilbert_exterior_angle_aux_parallel},$$

which proves

$$AB \parallel CE.$$

Finally,

hilbert_exterior_angle_less

combines the auxiliary construction, the extension $B - C - D$, and the parallelism to establish

$$\angle BAC < \angle ACD.$$

The Euclidean theorem itself therefore contains no unresolved construction. The substantial work has been moved into the Hilbert interface, where it can be reused independently of Proposition I.16.

6.2 The Second Opposite Interior Angle

For the second inequality, the proof first derives $A \neq C$ from the noncollinearity of A, B, C . Hilbert's order axiom then extends AC through C to a point F :

$$A - C - F.$$

The triangle is viewed in the order BAC . Applying `euclid_proposition_16_first` to this reordered triangle gives

$$\angle ABC < \angle BCF.$$

It remains to identify $\angle BCF$ with the desired exterior angle $\angle ACD$. From $B - C - D$ we obtain the reversed betweenness relation

$$D - C - B.$$

The two straight lines ACF and BCD therefore form a pair of vertical angles at C . The theorem `VerticalAngles` yields, after the required symmetry and reversal operations,

$$\angle BCF \cong \angle ACD.$$

The theorem

hilbert_angleLess_transport_right

then transports the comparison across the congruent right-hand angle:

$$\angle ABC < \angle BCF, \quad \angle BCF \cong \angle ACD \implies \angle ABC < \angle ACD.$$

7 Lean Formalization

The first half is formalized as follows:

```

theorem euclid_proposition_16_first
  [HilbertIncidence Geo]
  [HilbertCongruence Geo]
  (A B C D : Geo.Point)
  (hABC : not PrimCollinear Geo A B C)
  (hBCD : Geo.Between B C D) :
  HilbertAngleLess Geo B A C A C D := by

```

```

rcases
  hilbert_exterior_angle_aux
    Geo A B C hABC with
  <M, E,
    hAMC,
    hAMMC,
    hBME,
    hBMME,
    hAngle>

have hParallel :
  Geo.Parallel A B C E :=
  hilbert_exterior_angle_aux_parallel
    Geo A B C M E
    hABC
    hAMC
    hBME
    hAngle

exact
  hilbert_exterior_angle_less
    Geo A B C D M E
    hABC
    hAMC
    hBME
    hBCD
    hAngle
    hParallel

```

The second half deliberately reuses the first:

```

have hLessBCF :
  HilbertAngleLess Geo A B C B C F :=
  euclid_proposition_16_first
    Geo
    B A C F
    hBACnc
    hACF

```

The vertical-angle relation is then constructed at C and normalized to the required orientation:

```

have hVerticalRaw :
  Geo.AngleCongruent A C D F C B :=
  VerticalAngles
    Geo
    A C D F B
    hACF
    hDCA
    hACDnc

have hVerticalSymm :
  Geo.AngleCongruent F C B A C D :=
  Geo.angle_congruent_symmetry
    A C D
    F C B

```

```

hVerticalRaw

have hVertical :
  Geo.AngleCongruent B C F A C D :=
  (Geo.angle_congruent_reverse_first
   F C B A C D).mp
hVerticalSymm

```

Finally the strict inequality is transported:

```

exact
  hilbert_angleLess_transport_right
    Geo
    A B C
    B C F
    A C D
    hLessBCF
    hACDnc
    hVertical

```

The complete proposition simply combines the two parts:

```

theorem euclid_proposition_16
  [HilbertIncidence Geo]
  [HilbertCongruence Geo]
  (A B C D : Geo.Point)
  (hABC : not PrimCollinear Geo A B C)
  (hBCD : Geo.Between B C D) :
  HilbertAngleLess Geo B A C A C D /\
  HilbertAngleLess Geo A B C A C D := by

  have hFirst :=
    euclid_proposition_16_first
      Geo A B C D hABC hBCD

  have hSecond :=
    euclid_proposition_16_second
      Geo A B C D hABC hBCD

  exact <hFirst, hSecond>

```

8 Relationship with Earlier Propositions

8.1 Proposition I.10

Euclid's classical proof uses a midpoint construction as part of the auxiliary configuration. Proposition I.10 is therefore one of the constructional ingredients behind the traditional argument.

In the Hilbert reconstruction, however, the decisive exterior-angle comparison has already been isolated in reusable infrastructure, so the public I.16 proof does not need to reproduce the entire midpoint construction locally.

8.2 Proposition I.15

Vertical-angle congruence is used when the auxiliary configuration produces an angle in the opposite orientation from the one required in the final statement.

The theorem `VerticalAngles` therefore acts as an orientation bridge inside the reconstruction.

8.3 Transition to Proposition I.17

Proposition I.16 supplies the strict exterior-angle comparison that drives Proposition I.17. In I.17, each assertion that two interior angles are together less than two right angles is obtained by extending a side and applying the exterior-angle theorem in an appropriate orientation.

Thus I.16 begins the strict angle-order block of Book I that continues through I.17–I.19.

9 Hilbert and Book Zero Components Used

9.1 `HilbertAngleLess`

The proposition requires a genuinely synthetic notion of strict angle comparison. The relation `HilbertAngleLess` expresses that a congruent copy of one angle occurs as a proper interior subangle of another. This avoids introducing numerical angle measure and keeps Proposition I.16 inside pure incidence, order, and congruence geometry.

9.2 `hilbert_exterior_angle_aux`

This theorem supplies the auxiliary points used in the first exterior-angle construction. It packages the midpoint-style configuration that underlies Euclid’s original proof and returns the betweenness and congruence data required by the subsequent argument.

9.3 `hilbert_exterior_angle_aux_parallel`

From the auxiliary configuration, this theorem proves

$$AB \parallel CE.$$

The parallel relation is not an additional Euclidean postulate here; it is a derived relation inside the neutral Hilbert development used to organize the exterior-angle argument.

9.4 `hilbert_exterior_angle_less`

This is the principal interface theorem used by the first half of I.16. It converts the auxiliary construction and its parallelism into the strict angle comparison

$$\angle BAC < \angle ACD.$$

Its proof depends on the interior-subangle machinery developed in the Hilbert layer.

9.5 Interior-subangle transport

A substantial part of the work required for Proposition I.16 lies below the final Euclidean file. The Hilbert interface establishes that an interior ray can be transported between congruent angles while preserving the corresponding subangle.

The completed infrastructure includes the segment and crossbar mechanisms needed for that transport, in particular the ability to transport strict segment comparison through congruence, transport an interior point between congruent segments, and preserve ray–segment intersection when the endpoints are moved along the same two rays.

This is the part of the reconstruction where the formal proof exposes geometric structure that is largely implicit in the paper proof.

9.6 VerticalAngles

The second half uses the vertical-angle theorem already exposed by Proposition I.15. After AC is extended through C to F , the angles

$$\angle BCF \quad \text{and} \quad \angle ACD$$

are vertical. Their congruence allows the second inequality to be reduced to the first.

9.7 hilbert_angleLess_transport_right

This theorem expresses invariance of strict angle comparison under congruence of the right-hand angle. In the final step,

$$\angle ABC < \angle BCF$$

and

$$\angle BCF \cong \angle ACD$$

imply

$$\angle ABC < \angle ACD.$$

10 Formalization Notes

Proposition I.16 is a major transition from angle congruence to strict angle comparison.

Earlier propositions establish equalities of angles through triangle congruence, isosceles geometry, supplementary configurations, and vertical angles. I.16 requires a genuinely ordered conclusion: an exterior angle is strictly greater than either opposite interior angle.

The reconstruction therefore uses `HilbertAngleLess` rather than a numerical angle measure. The strict comparison is obtained from an explicit geometric subangle configuration and then transported through angle congruences until its endpoints match the Euclidean statement.

A second important feature is that the two opposite interior angles are not handled by pretending that orientation is irrelevant. One branch can be obtained directly from the exterior-angle infrastructure; the other requires additional transport, including vertical-angle congruence. Lean forces these orientations to be recorded explicitly.


Schematically, the formal mechanism is

auxiliary exterior-angle configuration $\longrightarrow$ strict subangle comparison $\longrightarrow$ angle-congruence transport $\longrightarrow$ I.16.

This infrastructure becomes immediately reusable. Proposition I.17 uses I.16 to control sums of two interior angles without introducing angle addition, while Propositions I.18 and I.19 use the resulting strict angle-order theory to relate side order and angle order in a triangle.

11 Dependency Summary

The formal dependency structure can be summarized as



The final Euclidean theorem is therefore compact, but its compactness is meaningful: the difficult geometry has been factored into reusable Hilbert-interface results rather than hidden in an axiom or a local `sorry`.

12 Conclusion

Proposition I.16 marks a significant step beyond the immediately preceding propositions. It introduces a strict comparison of angles and requires a robust synthetic notion of interiority. In Euclid’s presentation this is handled by a midpoint construction, SAS, and the observation that one angle is a proper part of another. In the formal Hilbert reconstruction the same content is decomposed into explicit reusable mechanisms for rays, betweenness, congruence, crossbar behavior, and transport of strict angle comparison.

The final Lean theorem is short precisely because those mechanisms have been established independently. No additional axiom is introduced in `Proposition16.lean`, and the completed development compiles without `sorry`.