

Euclid's Elements — Book I

Proposition 35

Jurek Rózański
Military University of Technology, Warsaw, Poland
`jerzy.rozanski@wat.edu.pl`

CGJTEAMLAB
Formal Reconstruction — Technical Documentation

Parallelograms which are on the same base and in the same parallels equal one another.

1 Introduction

Euclid's Proposition I.35 is the first proposition in Book I in which the word “equal” no longer refers merely to congruent segments or congruent angles. The proposition states that two parallelograms on the same base and in the same parallels are equal to one another. The two parallelograms need not be congruent. What Euclid uses is a cut-and-paste comparison of rectilinear figures.

This makes I.35 a genuine transition point in the reconstruction. Up to I.34, the main formal objects are points, lines, betweenness, segment congruence, angle congruence, and parallelism. At I.35 one must decide what “equal figures” means before the classical argument can even be stated faithfully.

The present development does not introduce numerical area. Instead it introduces a small formal scissors calculus. Triangles are the basic pieces, finite formal sums of triangles represent decompositions, and a generated relation records the operations of splitting a triangle, replacing it by a congruent triangle, adding decompositions, and composing such steps.

The final Lean theorem therefore concludes not a numerical area equality but formal equicomplementability:

```
theorem euclid_proposition_35
  [HilbertIncidence Geo]
  [HilbertEuclideanPlane Geo]
  (A B C D E F : Geo.Point)
  (hABCD : IsParallelogram Geo A B C D)
  (hEBCF : IsParallelogram Geo E B C F)
  (hADF : Geo.Between A D F)
  (hAEF : Collinear Geo A E F) :
  HilbertScissorsEquicomplementable Geo
    (hilbertParallelogramTerm Geo A B C D)
    (hilbertParallelogramTerm Geo E B C F)
```

The extra order and collinearity hypotheses are not cosmetic. They make explicit part of the configuration which is left implicit by the classical diagram and are essential to a complete formal case analysis.

2 The Classical Configuration

Let $ABCD$ and $EBCF$ be parallelograms on the common base BC . Their opposite sides AD and EF lie on one straight line which is parallel to BC . Euclid's statement is therefore

$$ABCD \text{ and } EBCF \text{ on the same base } BC \implies ABCD = EBCF$$

in Euclid's language of equality of rectilinear figures.

A complete synthetic proof has to distinguish three possible positions of the two upper sides. Wilkins isolates exactly these configurations:

1. the upper sides overlap, with order A, E, D, F ;
2. the upper sides meet at a common endpoint, so $E = D$;
3. the upper sides are disjoint, with order A, D, E, F .

Only the third configuration is treated explicitly in Euclid's proof. The first two are not exceptional from the formal point of view: they are genuine cases which a theorem prover cannot recover from the picture.




Figure 1: The three upper-side configurations relevant to Proposition I.35. The point G is required only in the disjoint case.

The current Lean theorem uses $A - D - F$ as an outer container:

$$A - D - F, \quad A, E, F \text{ collinear.}$$

From the parallelogram hypotheses and this data, the reconstruction first proves $A - E - F$. The relative order of the two interior points D and E then yields precisely the three cases shown in Figure 1.

3 Classical Proof

Euclid's explicit proof is the disjoint case. Since $ABCD$ and $EBCF$ are parallelograms, Proposition I.34 gives

$$AD = BC = EF, \quad AB = DC.$$

Because DE is common to the two longer upper segments, Common Notions 1 and 2 yield

$$AE = DF.$$

The lines AB and DC are parallel, and the upper line AF is a transversal. Proposition I.29 therefore gives

$$\angle EAB = \angle FDC.$$

Euclid then uses Proposition I.4 to conclude that the outer triangles are equal:

$$\triangle EAB = \triangle FDC.$$

In the disjoint configuration the lines EB and DC meet at G . Euclid subtracts the common triangle DGE from the two equal outer triangles, obtaining equal trapezia, and then adds the common triangle GBC . This produces the two parallelograms.

The classical dependency graph is therefore

$$\begin{array}{c}
\text{I.34} + \text{I.29} \\
\Downarrow \\
AE = DF, \quad AB = DC, \quad \angle EAB = \angle FDC \\
\Downarrow \quad \text{I.4} \\
\triangle EAB = \triangle FDC \\
\Downarrow \quad \text{C.N.3} \\
\text{equal remainders} \\
\Downarrow \quad \text{C.N.2} \\
ABCD = EBCF.
\end{array}$$

The overlap and common-endpoint cases use the same outer-triangle comparison but a different addition pattern. Wilkins' three-case analysis is therefore a useful completion of Euclid's single drawn configuration.

4 Equidecomposition and Equicomplementability

The decisive modern control source for I.35 is Hilbert's theory of plane area. In Chapter IV, Section 18 of the *Foundations of Geometry*, Hilbert defines two polygons to be equidecomposable when they can be triangulated into finitely many pairwise congruent triangles. He then defines equicomplementability by allowing pairs of equidecomposable polygons to be adjoined before comparing the resulting figures by equidecomposition.

In Section 19 Hilbert states Theorem 44: two parallelograms with the same bases and the same altitudes are equicomplementable. This is the Hilbertian counterpart of the Euclidean area argument which begins at I.35.

The distinction between the two notions is essential. Equidecomposability is a direct finite scissors congruence. Equicomplementability is weaker: the figures may become equidecomposable only after suitable complements have been adjoined. Hilbert explicitly notes that in non-Archimedean geometry figures with equal bases and altitudes may be equicomplementable without being equidecomposable.

Hartshorne makes the same distinction under the name *equal content*. His definition requires figures P, Q and P', Q' to be nonoverlapping, requires Q and Q' to be equidecomposable, and requires the two enlarged figures to be equidecomposable. His Example 22.1.2 identifies Euclid I.35 as an example of equal content.

This matters for the present Lean development because the formal calculus is not yet a complete model of polygonal regions. It records formal sums of triangles, but it does not yet carry an interior or a nonoverlap predicate. Consequently

`HilbertScissorsEquicomplementable`

should be read exactly as its source comment says: a *term-level analogue* of Hilbert’s equicomplementability relation. The theorem proved in I.35 is therefore a faithful algebraic scissors certificate, but a later semantic layer would still be needed to identify every formal sum with an actual nonoverlapping polygonal union.

4.1 Why numerical area is deliberately avoided

A modern proof could assign an area to each parallelogram and write

$$\text{area} = \text{base} \cdot \text{height}.$$

That would be the wrong logical direction here. Numerical area is developed later than the elementary scissors theory in both Hilbert and Hartshorne. Using it at I.35 would replace Euclid’s synthetic argument by a stronger and logically downstream theorem.

The present reconstruction therefore stays at the level of finite cutting, congruence, and recombination.

5 The Hilbert Scissors Calculus

The file `HilbertScissors.lean` introduces two formal types:

```
abbrev HilbertTriangleTerm :=
  Multiset Geo.Point

abbrev HilbertScissorsTerm :=
  Multiset (HilbertTriangleTerm Geo)
```

A triangle is represented by an unordered multiset of its three vertices. A scissors term is an unordered multiset of triangle terms. This choice has two useful formal consequences:

- the order of the three names used to write a triangle is not geometric data;

- the order in which the triangular pieces are listed is not geometric data either.

Thus much of the bookkeeping which would otherwise be expressed by explicit list permutations disappears at the representation level.

The central generated relation is

```

inductive HilbertScissorsEq
  (Geo : Geometry.Geo) :
  HilbertScissorsTerm Geo ->
  HilbertScissorsTerm Geo ->
  Prop

| refl
  (P : HilbertScissorsTerm Geo) :
  HilbertScissorsEq Geo P P

| symm
  {P Q : HilbertScissorsTerm Geo}
  (h : HilbertScissorsEq Geo P Q) :
  HilbertScissorsEq Geo Q P

| trans
  {P Q R : HilbertScissorsTerm Geo}
  (hPQ : HilbertScissorsEq Geo P Q)
  (hQR : HilbertScissorsEq Geo Q R) :
  HilbertScissorsEq Geo P R

| add
  {P Q R S : HilbertScissorsTerm Geo}
  (hPQ : HilbertScissorsEq Geo P Q)
  (hRS : HilbertScissorsEq Geo R S) :
  HilbertScissorsEq Geo
    (P + R)
    (Q + S)

| split
  (A B C X : Geo.Point)
  (hBXC : Geo.Between B X C) :
  HilbertScissorsEq Geo
    (hilbertScissorsTriangle Geo A B C)
    (hilbertScissorsTriangle Geo A B X +
     hilbertScissorsTriangle Geo A X C)

| congruent
  (A B C D E F : Geo.Point)
  (hCong :
    TriangleCongruenceResult
      Geo A B C D E F) :
  HilbertScissorsEq Geo
    (hilbertScissorsTriangle Geo A B C)
    (hilbertScissorsTriangle Geo D E F)

```

The two geometric generators are therefore exactly the expected ones:

- **split**: cut a triangle at a point lying between two vertices;

- **congruent**: replace one triangle by a congruent triangle.

The remaining constructors make this relation an equivalence relation compatible with formal addition.

For formal equicomplementability, the implementation asks for formal complements R and S such that

$$R \sim S \quad \text{and} \quad P + R \sim Q + S,$$

where $\sim$ denotes **HilbertScissorsEq**. This is the algebraic pattern of Hilbert's definition, while the geometric nonoverlap conditions remain outside the present term language.

5.1 Triangulation lemmas

Two reusable lemmas are particularly important:

```
parallelogram_two_triangulations
crossing_quadrilateral_two_triangulations
```

The first proves that the two diagonal triangulations of a parallelogram are scissors-equivalent. It uses the existing theorem that the diagonals of a Euclidean parallelogram meet internally.

The second is more general. If the diagonals AC and BD of a quadrilateral meet internally at M , then

$$ABC + ACD \sim ABD + BCD.$$

The scissors move is easier to read as a three-stage refinement.

$$ABC + ACD$$




Figure 2: First triangulation of the quadrilateral, along the diagonal AC .

First, split the two triangles along the intersection point M of the diagonals. This refines the decomposition into the same four elementary triangles.

$$ABM + BMC + CMD + DMA$$




Figure 3: Common refinement at the diagonal intersection M .

$$ABD + BCD$$




Figure 4: Second triangulation of the same quadrilateral, along the diagonal BD .

The four pieces can then be regrouped along the other diagonal BD .

Its proof is pure scissors algebra after the two betweenness witnesses are provided. This lemma is what makes the overlap case short: the common trapezoid $EBCD$ can be flipped from one diagonal triangulation to the other without introducing a special area theorem.

6 Proof Architecture of the Reconstruction

The actual Lean proof is not a mechanical transcription of Euclid's proof. It has two logically distinct stages.

6.1 Stage 1: recover the upper-line order

The final theorem is given

$$A - D - F \quad \text{and} \quad A, E, F \text{ collinear.}$$

The theorem `i35_upper_order` proves

$$A - E - F.$$

The proof uses opposite-side congruence for the two parallelograms to obtain $AD \cong EF$. Hilbert's betweenness trichotomy on A, E, F then gives three possible orders. The two bad orders are excluded separately.

The first exclusion,

$$\text{i35_not_EAF,}$$

is a segment-order argument. From $A - D - F$ one gets $AD < AF$; congruence transports this to $EF < AF$. The bad order $E - A - F$ would imply the reverse strict inequality, contradicting asymmetry.

The second exclusion,

$$\text{i35_not_AFE,}$$

is genuinely planar. The helper `i35_cross_intersection` first constructs a point where BF crosses AC . The diagonal theorem for the second parallelogram gives another point where BF crosses EC . Under the bad order $A - F - E$, the same line would also meet the third side AE of the triangle AEC . The line-separation theorem `hilbert_line_avoids_third_triangle_side` forbids one line from meeting all three sides in the required interior configuration.

Once $A - E - F$ is known, the pure order theorem `i35_right_cases_of_common_container` compares the two interior points D and E and produces exactly

$$A - E - D - F, \quad E = D, \quad \text{or} \quad A - D - E - F.$$

6.2 Stage 2: solve the three scissors cases

The three cases are dispatched by `i35_from_three_cases`.

6.2.1 Overlap case

The outer triangles EAB and FDC are shown congruent. The common trapezoid $EBCD$ is then re-triangulated by `crossing_quadrilateral_two_triangulations`. No nonzero complement is needed. The two parallelogram terms are directly scissors-equivalent, so they are equicomplementable with zero complements.

6.2.2 Common-endpoint case

The two parallelograms share the upper endpoint. Again no auxiliary point G is required. The proof changes the triangulations and replaces the two outer triangles by congruent ones. The

resulting parallelogram terms are directly scissors-equivalent, hence equicomplementable with zero complements.

6.2.3 Disjoint case

This is Euclid's explicit configuration. Here the point G is essential. The helper

`i35_disjoint_intersection`

constructs G with

$$E - G - B, \quad D - G - C.$$

The existence proof is not read off from the diagram. It uses Hilbert order and Pasch-type intersection machinery, followed by parallel/order transport to place the intersection on both required segments.

Instead of literally subtracting DGE as Euclid does, the formal proof uses the equivalent completion pattern. It is useful to display the four stages separately.




Figure 5: Step 1. Adjoin the triangle DGE to the parallelogram $ABCD$.

Using the betweenness relations $E - G - B$ and $D - G - C$, split and regroup the augmented figure. The result is the outer triangle EAB together with the common triangle GBC .




Figure 6: Step 2. The augmented left-hand figure is rearranged as $EAB + GBC$.

The outer triangles EAB and FDC are congruent. The scissors calculus therefore replaces the first outer triangle by the second while leaving the common triangle GBC unchanged.

Finally, reverse the splitting and regrouping on the right-hand side.




Figure 7: Step 3. Replace the congruent outer triangle: $EAB + GBC \sim FDC + GBC$.




Figure 8: Step 4. Regroup the pieces as the second augmented parallelogram $EBCF + DGE$.

Thus the complete formal chain is

$$\begin{aligned}
 ABCD + DGE &\sim EAB + GBC \\
 &\sim FDC + GBC \\
 &\sim EBCF + DGE.
 \end{aligned}$$

This is exactly the formal equicomplementability witness needed for I.35.

6.3 The complete Lean DAG

The reconstruction can be summarized as

two parallelograms on $BC + A - D - F + \text{Collinear}(A, E, F)$
 $\Downarrow$
 $AD \cong EF + \text{segment order} + \text{line separation}$
 $\Downarrow$
 $A - E - F$
 $\Downarrow$
 $\boxed{A - E - D - F} \quad \boxed{E = D} \quad \boxed{A - D - E - F}$
 $\Downarrow$
 overlap scissors proof common-endpoint proof disjoint proof with G
 $\Downarrow$
 HilbertScissorsEquicomplementable
 $\Downarrow$
 euclid_proposition_35.

7 Lean Formalization

The final Proposition file is intentionally short. It imports the separate scissors module and exposes only the Book I theorem:

```

import CGJteamLab.HilbertScissors

namespace Geometry

universe u

variable (Geo : Geometry.Geo)

theorem euclid_proposition_35
  [HilbertIncidence Geo]
  [HilbertEuclideanPlane Geo]
  (A B C D E F : Geo.Point)
  (hABCD : IsParallelogram Geo A B C D)
  (hEBCF : IsParallelogram Geo E B C F)
  (hADF : Geo.Between A D F)
  (hAEF : Collinear Geo A E F) :
  HilbertScissorsEquicomplementable Geo
    (hilbertParallelogramTerm Geo A B C D)
    (hilbertParallelogramTerm Geo E B C F) := by
exact
  i35_complete
  Geo
  A B C D E F
  hABCD
  hEBCF
  hADF
  hAEF

end Geometry

```

All substantial work is therefore isolated in `HilbertScissors.lean`. This is deliberate. Proposi-

tion I.35 is the first client of the scissors language, but the formal content calculus is not part of the statement of Euclid I.35 itself.

The endpoint theorem `i35_complete` is also thin. Its role is to package the established ordered proof:

$$\begin{array}{c}
 \text{i35_upper_order} \\
 \Downarrow \\
 \text{i35_right_cases_of_common_container} \\
 \Downarrow \\
 \text{i35_from_three_cases} \\
 \Downarrow \\
 \text{i35_complete.}
 \end{array}$$

8 Relationship with Earlier Propositions

8.1 Proposition I.29

Euclid uses I.29 explicitly to identify the corresponding angles $\angle EAB$ and $\angle FDC$. The current Lean proof does not replay this exact SAS route at the top level. Parallelism enters through the developed Hilbert theory and through the parallelogram infrastructure.

8.2 Proposition I.34

I.34 is the immediate metric input. Opposite sides of the two parallelograms are congruent, so

$$AD \cong BC \cong EF$$

and the second pair of opposite sides provides the remaining outer-triangle side congruence. The Lean helper `ParallelogramOppositeSidesCongruent` packages this information.

The current reconstruction then proves the outer triangles by the already available Hilbert SSS theorem rather than by Euclid's SAS argument. This is another example of the project deliberately reusing a stronger developed Hilbert API rather than mechanically replaying the historical DAG.

8.3 Transition to Proposition I.36

I.36 is a direct generalization from the same base to equal bases. Euclid's classical proof uses I.33 to construct an intermediate parallelogram and then applies I.35 twice. For the formal development this means that I.35 is not an isolated experiment: the scissors/equicomplementability language is now a real dependency of the next area propositions.

9 Hilbert and Book Zero Components Used

The I.35 reconstruction uses a much broader part of the developed Hilbert interface than the final 38-line Proposition file suggests.

- `IsParallelogram` and the Euclidean parallelogram theory;
- `ParallelogramOppositeSidesCongruent`;
- `ParallelogramDiagonalIntersectionExists`;
- the derived `HilbertSSS` triangle-congruence theorem;
- Hilbert betweenness trichotomy and inner/outer transitivity;
- synthetic segment comparison and transport under congruence;
- Pasch/intersection machinery for the disjoint case;
- line separation through `hilbert_line_avoids_third_triangle_side`;
- transport of betweenness from collinearity and parallelism;
- the newly introduced formal scissors layer in `HilbertScissors.lean`.

No new Book Zero theorem is introduced by the final Proposition file. The new reusable material is instead concentrated in the scissors module.

The source hierarchy also changes emphasis at this point. Borsuk remains a useful control source for order, congruence, and parallelogram geometry, but the new notion of equality of figures is controlled more directly by Hilbert’s Chapter IV and Hartshorne’s theory of content. The Beeson proof corpus remains useful primarily as an audit of hidden incidence and order conditions rather than as the architecture of the Lean proof.

10 Formalization Notes

10.1 Axiomatic status

The final theorem assumes

```
[HilbertEuclideanPlane Geo]
```

and is therefore Euclidean in the current project API. This is consistent with Hilbert’s Chapter IV treatment, which assumes the parallel axiom in addition to incidence, order, and congruence. The present file does not claim a sharper minimal axiom boundary for every internal helper.

The Euclidean strength is already visible in the reusable parallelogram results. Opposite-side congruence and the developed intersection/parallel machinery are called through the Euclidean-plane instance used by the project. Once the relevant congruences and betweenness data are available, individual scissors rewrites themselves are purely formal.

10.2 Hidden configuration data

I.35 exposes more hidden diagrammatic information than the immediately preceding propositions. The formal proof must explicitly manage:

- the outer order $A - D - F$;

- collinearity of A, E, F ;
- the derived order $A - E - F$;
- the trichotomy between overlap, common endpoint, and disjoint upper sides;
- existence and exact betweenness position of the point G in the disjoint case;
- internal diagonal intersections used for re-triangulation;
- non-collinearity needed for Pasch and line-separation arguments.

The point G is especially instructive. It is not a universal witness for all forms of I.35. It appears only in the disjoint configuration. Trying to force one G -based proof onto every configuration would obscure the geometry and create unnecessary order obligations.

10.3 The scissors language is formal, not yet semantic

The nested `Multiset` representation is a formal algebra of triangle terms. It intentionally forgets vertex order and piece order, but it also forgets geometric interiors. Therefore formal addition is total even when two actual geometric pieces would overlap.

This is the main semantic limitation of the present module. A future polygonal semantics could add a nonoverlap predicate, partial composition, or a realization theorem connecting valid formal scissors derivations to nonoverlapping rectilinear figures. None of that is required to complete the current I.35 certificate, but the distinction is documented here so that the formal relation is not silently identified with a stronger geometric notion.

10.4 Reusable helpers introduced

The most obviously reusable scissors helpers are

```
triangle_swap12
triangle_swap23
triangle_cycle
scissors_triangle_swap12
scissors_triangle_swap23
scissors_triangle_cycle
parallelogram_two_triangulations
crossing_quadrilateral_two_triangulations
equicomplementable_refl
equicomplementable_symm
```


The remaining helpers whose names begin with `i35_` are deliberately left proposition-specific for now. Some may later reveal a more general Hilbert statement, but no architectural promotion is required for I.35.

11 Dependency Summary

The classical dependency is

$$\text{I.34} + \text{I.29} + \text{I.4} + \text{Common Notions 1-3} \longrightarrow \text{I.35.}$$

The actual Lean dependency is better represented as



New axiom:	no
New formal framework:	yes — HilbertScissors.lean
New Book Zero helper:	no
Build clean:	yes
sorry:	no

12 Conclusion

Proposition I.35 is a structural turning point in the formal reconstruction of Book I. The geometry is still elementary, but the proposition introduces an equality of figures which cannot be represented faithfully by segment or angle congruence alone.

Hilbert’s distinction between equidecomposability and equicomplementability provides the correct conceptual guide. The present Lean development follows that guide by introducing a formal scissors calculus rather than a numerical area function. The calculus is intentionally modest: it is a term-level algebra of triangular decompositions, not yet a full semantic theory of polygonal regions.

The proof itself also exposes a classical omission which is invisible on the usual diagram. There are three upper-side configurations, and the auxiliary intersection point G belongs only to the disjoint one. Once this case split is made at the correct level, the proof decomposes naturally into order geometry, triangle congruence, and formal scissors recombination.

The resulting Proposition file is short because the complexity has been placed in a reusable and explicitly named layer. This is exactly the kind of separation needed for the next area propositions, beginning with I.36.