Euclid's Elements — Book I
Proposition 20

CGJteamLab

In any triangle two sides taken together in any manner are greater than the remaining one.

1 Introduction

Euclid's Proposition I.20 is the triangle inequality in purely synthetic form. For a nondegenerate triangle ABC, Euclid proves that any two sides taken together are greater than the third. For one of the three cyclic cases the statement is

$$BA + AC > BC.$$

The Hilbert reconstruction does not introduce a numerical length function and does not define an arithmetic sum of real numbers. Instead, the sum $BA + AC$ is represented geometrically. Extend BA beyond A to a point D such that

$$B - A - D \qquad \text{and} \qquad AD \cong AC.$$

Then the whole segment BD represents the concatenation of BA with a copy of AC. The required inequality becomes

$$BC < BD.$$

Thus the formal theorem asserts the existence of such a point D together with the strict segment comparison $BC < BD$.

The completed proof remains entirely inside neutral Hilbert geometry. It uses incidence, order, congruence, Proposition I.5, the synthetic angle-order relation, and Proposition I.19. No parallel postulate, numerical angle measure, or numerical segment length is used.

2 The Classical Configuration

Let ABC be a nondegenerate triangle. Extend BA beyond A to a point D and choose D so that

$$AD \cong AC.$$

The construction records

$$B - A - D, \qquad AD \cong AC.$$

The target is

$$BC < BD.$$

Geometrically, this is exactly the inequality

$$BC < BA + AC.$$



Figure 1: Classical configuration for Proposition I.20. The side BA is produced beyond A to D, with $AD \cong AC$.

3 Classical Proof

After extending BA to D so that $AD = AC$, join CD.

The auxiliary triangle ADC is isosceles. Proposition I.5 therefore gives

$$\angle ADC = \angle ACD.$$

Since A lies between B and D, the ray CA lies inside the angle BCD. Hence

$$\angle ACD < \angle BCD.$$

Combining this with the isosceles-triangle equality gives

$$\angle BDC < \angle BCD.$$

Now apply Proposition I.19 to triangle BCD. The greater angle is opposite the greater side, so

$$BC < BD.$$

Finally, because $B - A - D$ and $AD = AC$, the segment BD is the geometric concatenation of BA and AC. Therefore

$$BA + AC > BC.$$

The classical dependency pattern is

$$\begin{array}{c} B - A - D,\ AD \cong AC \\ \downarrow \\ \angle ADC \cong \angle ACD \quad \text{(I.5)} \\ \downarrow \\ \angle BDC < \angle BCD \\ \downarrow \quad \text{(I.19)} \\ BC < BD. \end{array}$$

The single diagram above already contains every geometric ingredient needed for this proof. There is no case split to visualize. In particular, the important graphical information is not merely the shape of triangle BCD, but the collinear order $B - A - D$ together with the marked equality $AD \cong AC$. These two facts explain why BD represents the synthetic sum $BA + AC$.

Wilkins also records an ancient alternative proof, attributed by Proclus to Heron or Porphyry, which bisects $\angle BAC$ and uses I.16 and I.19. That proof has a genuinely different diagram, but it is not the proof formalized here, so it is not added to the main reconstruction.

4 Synthetic Form of the Triangle Inequality

The formal reconstruction deliberately avoids introducing an arithmetic addition operation on segments. The public result is expressed by an extension point D:

$$\exists D\ (B - A - D\ \wedge\ AD \cong AC\ \wedge\ BC < BD).$$

This is the natural Hilbert form of Euclid's phrase "two sides taken together are greater than the remaining one."

The representation has two advantages. First, it remains purely geometric. Second, it makes explicit which part of the statement belongs to order and congruence geometry: the "sum" is a concatenated segment, not an externally imposed numerical quantity.

5 Proof Architecture of the Reconstruction

The formal proof follows Euclid's geometric strategy closely, but it separates the argument into reusable logical stages. An extension point turns a sum of two sides into a single segment; an interior-angle comparison and an isosceles-triangle transport produce the angle inequality required by Proposition I.19; finally the resulting strict segment comparison is transported back to the original sides.

```
triangle ABC
    |
extend BA beyond A
    |
construct D with B-A-D and AD ~= AC
    |
    +------------------------------+
    |                              |
    v                              v
BD represents BA+AC         triangle ACD is isosceles
                                   |
                                  I.5
                                   |
                           ADC ~= ACD
                                   |
                    CA lies inside angle DCB
                                   |
                           ACD < DCB
                                   |
                       congruence transport
```

```
                                       |
                                 BDC < BCD
                                       |
                                      I.19
                                       |
                                    BC < BD
    |                                  |
    +--------------+---------------+
                   |
                   v
             BC < BA+AC
```

Unlike I.21–I.24, no second geometric picture is needed here: the formal architecture is a refinement of the same Euclidean configuration rather than a different case structure or a stronger primitive construction.

5.1 Construction of the Extension Point

The point D is constructed in two stages.

First, Hilbert's order axiom extends the line beyond A:

$$B - A - R.$$

In Lean this is obtained from

`HilbertOrder.between_extension`.

Second, a copy of AC is laid off from A on the ray AR using

`HilbertCongruence.segment_construction`.

This produces a point D satisfying

$$AD \cong AC$$

and placing D on the same ray from A as R.

The theorem

`hilbert_between_transport_sameRays`

then transports the original betweenness relation $B - A - R$ to

$$B - A - D.$$

Thus the geometric "sum" segment BD is obtained entirely from existing order and congruence infrastructure.

5.2 The Interior-Angle Step

The key order-theoretic observation is that the ray CA meets the open segment DB at the point A itself. Since

$$D - A - B,$$

the witness for

```
HilbertRayMeetsSegment Geo C A D B
```

is simply A.

The strict angle comparison is then introduced directly with

`hilbert_angleLess_intro`.

The first angle is $\angle ACD$. The corresponding proper subangle of $\angle DCB$ is represented by $\angle DCA$, which is the same unoriented angle as $\angle ACD$. The representation-level symmetry of angles is supplied by

`angle_congruent_reverse_second`

together with reflexivity of angle congruence.

This yields

$$\angle ACD < \angle DCB.$$

No auxiliary half-plane or perpendicular construction is required. The entire strict comparison follows from the ray meeting the opposite segment.

5.3 Transport Through the Isosceles Triangle

From

$$AD \cong AC$$

and the noncollinearity of A, D, C, Proposition I.5 is available in the form

`hilbert_isosceles_base_angles`.

It gives

$$\angle ADC \cong \angle ACD.$$

For Proposition I.20 we also introduced the reusable theorem

`hilbert_angleLess_transport_left`.

It expresses invariance of strict angle comparison under replacing the smaller angle by a congruent angle.

Applied here, it transports

$$\angle ACD < \angle DCB$$

across

$$\angle ADC \cong \angle ACD,$$

producing

$$\angle ADC < \angle DCB.$$

Since $B - A - D$, the rays DA and DB coincide. The theorem

`hilbert_angle_eq_of_sameRay_first`

therefore identifies the angle $\angle ADC$ with $\angle BDC$. Hence

$$\angle BDC < \angle DCB.$$

Finally, reversal of the two boundary rays at C identifies

$$\angle DCB \cong \angle BCD.$$

The existing theorem

`hilbert_angleLess_transport_right`

then gives

$$\angle BDC < \angle BCD.$$

5.4 Application of Proposition I.19

At this point the proof has produced exactly the hypothesis needed for Proposition I.19 in triangle BCD:

$$\angle BDC < \angle BCD.$$

Applying

`euclid_proposition_19`

yields

$$BC < BD.$$

This is the synthetic triangle inequality for the chosen pair of sides. The other two cyclic cases follow by renaming the vertices.

6 Lean Formalization

The public theorem is:

```
theorem euclid_proposition_20
    [HilbertIncidence Geo]
    [HilbertCongruence Geo]
    (A B C : Geo.Point)
    (hABC : not PrimCollinear Geo A B C) :
    exists D : Geo.Point,
      Geo.Between B A D /\
      Geo.Congruent A D A C /\
      HilbertSegmentLess Geo B C B D := by
```

The first stage derives $A \neq B$ and extends BA beyond A:

```
have hAB : A != B :=
  hilbert_noncollinear_ne_first
    Geo A B C hABC

have hBA : B != A :=
  hAB.symm

rcases
    HilbertOrder.between_extension
      B A hBA with
  <R, hBAR>
```

A copy of AC is then laid off on the ray AR:

```
have hAR : A != R :=
  (HilbertOrder.between_incidence
    B A R hBAR).2.1

rcases
    HilbertCongruence.segment_construction
      (Geo := Geo)
      A C
      A R
      hAR with
  <D, hRayARD, hAD_AC>
```

Betweenness is transported to the actual constructed point D:

```
have hRayABA :
    HilbertSameRay Geo A B B :=
  hilbert_sameRay_refl
    Geo A B hBA

have hBAD :
    Geo.Between B A D :=
  hilbert_between_transport_sameRays
    Geo
    B A R
    B D
    hBAR
    hRayABA
    hRayARD
```

The theorem then packages the construction and continues with the strict segment comparison:

```
refine <D, hBAD, hAD_AC, ?_>
```

The interior ray is represented directly by the point A:

```
have hDAB :
    Geo.Between D A B :=
  (HilbertOrder.between_incidence
    B A D hBAD).2.2.2.2

have hRayCAA :
    HilbertSameRay Geo C A A :=
  hilbert_sameRay_refl
```

```
    Geo C A hAC

have hInside :
    HilbertRayMeetsSegment Geo C A D B :=
  <A, hDAB, hRayCAA>
```

Angle reversal is obtained through the representation-level congruence laws:

```
have hAngleRefl :
    Geo.AngleCongruent A C D A C D :=
  Geometry.Geo.angle_congruent_reflexive
    Geo A C D

have hAngleSwap :
    Geo.AngleCongruent A C D D C A :=
  (Geo.angle_congruent_reverse_second
    A C D
    A C D).mp
    hAngleRefl
```

This gives the strict subangle relation:

```
have hLessACD_DCB :
    HilbertAngleLess Geo A C D D C B :=
  hilbert_angleLess_intro
    Geo
    A C D
    D C B
    A
    hACD
    hDCB
    hInside
    hAngleSwap
```

Proposition I.5 and left transport give the corresponding comparison with the base angle at D:

```
have hIso :
    Geo.AngleCongruent A D C A C D :=
  hilbert_isosceles_base_angles
    Geo
    A D C
    hADC
    hAD_AC

have hLessADC_DCB :
    HilbertAngleLess Geo A D C D C B :=
  hilbert_angleLess_transport_left
    Geo
    A C D
    A D C
    D C B
    hLessACD_DCB
    hADC
    hIso
```

The ray DA is then replaced by the same ray DB:

```
have hRayDAB :
```

```
    HilbertSameRay Geo D A B :=
  hilbert_sameRay_of_between
    Geo D A B hDAB

have hRayDBA :
    HilbertSameRay Geo D B A :=
  hilbert_sameRay_symm
    Geo D A B hRayDAB

have hBDC_ACD :
    Geo.AngleCongruent B D C A C D := by
  unfold Geometry.Geo.AngleCongruent at hIso |-
  rw [hilbert_angle_eq_of_sameRay_first
    Geo D B A C hRayDBA]
  exact hIso
```

Using left transport again gives

```
have hLessBDC_DCB :
    HilbertAngleLess Geo B D C D C B :=
  hilbert_angleLess_transport_left
    Geo
    A C D
    B D C
    D C B
    hLessACD_DCB
    hBDC
    hBDC_ACD
```

Finally, the right-hand angle is reversed and Proposition I.19 closes the proof:

```
have hAngleReflDCB :
    Geo.AngleCongruent D C B D C B :=
  Geometry.Geo.angle_congruent_reflexive
    Geo D C B

have hDCB_BCD :
    Geo.AngleCongruent D C B B C D :=
  (Geo.angle_congruent_reverse_second
    D C B
    D C B).mp
    hAngleReflDCB

have hLessBDC_BCD :
    HilbertAngleLess Geo B D C B C D :=
  hilbert_angleLess_transport_right
    Geo
    B D C
    D C B
    B C D
    hLessBDC_DCB
    hBCD
    hDCB_BCD

exact
  euclid_proposition_19
    Geo
    B C D
```

```
    hBCD
    hLessBDC_BCD
```

7 Relationship with Earlier Propositions

7.1 Proposition I.5

The construction creates an isosceles triangle. Its base-angle congruence is the mechanism that transports the relevant angle from the auxiliary configuration back toward the original triangle.

7.2 Proposition I.16

The exterior-angle theorem supplies the strict angle comparison needed inside the auxiliary triangle. This is the order-theoretic part of the argument: the diagram alone is not used to infer that one angle is larger than another.

7.3 Proposition I.19

Proposition I.19 converts the strict angle comparison into a strict comparison of the opposite sides. In this sense I.20 is not an independent segment-order argument: its decisive inequality is obtained through the angle-order theory developed immediately before it.

7.4 Transition to Proposition I.21

Proposition I.20 proves the triangle inequality by representing the sum of two sides as a single constructed segment. Proposition I.21 develops this same synthetic segment-arithmetic viewpoint further. Thus I.20 is the first clear bridge from the angle-comparison block I.16–I.19 to the more explicit segment arithmetic used in I.21 and I.22.

8 Hilbert and Book Zero Components Used

Proposition I.20 required one new general theorem in the Hilbert interface:

`hilbert_angleLess_transport_left`.

Its role is symmetric to the previously available

`hilbert_angleLess_transport_right`.

Together they express that strict synthetic angle comparison is invariant under congruent replacement of either compared angle. This is not specific to the triangle inequality and is expected to be reusable in later propositions.

The proof also reuses the existing construction pattern

`between_extension` $\longrightarrow$ `segment_construction` $\longrightarrow$ `between_transport_sameRays`,

which already occurs elsewhere in the Hilbert development.

9 Dependency Summary

The formal dependency structure of Proposition I.20 can be summarized as

$$\begin{array}{c}
\text{Hilbert incidence + order + congruence} \\
\downarrow \\
\text{between extension + segment construction} \\
\downarrow \\
B - A - D, \quad AD \cong AC \\
\downarrow \\
\text{I.5 + HilbertAngleLess} \\
\downarrow \\
\angle BDC < \angle BCD \\
\downarrow \quad \text{I.19} \\
BC < BD \\
\downarrow \\
BA + AC > BC \quad \text{(synthetic interpretation).}
\end{array}$$

No Euclidean parallel axiom appears in this chain. Proposition I.20 is therefore a theorem of neutral geometry in the present Hilbert reconstruction.

10 Conclusion

Proposition I.20 is the first place in this sequence where Euclid's language of adding lengths is naturally reconstructed without introducing numerical length. The formal statement replaces the expression $BA + AC$ by an explicit extension segment BD satisfying

$$B - A - D, \qquad AD \cong AC.$$

The rest of the argument is purely synthetic. Proposition I.5 converts the copied side into an angle congruence, the interior-ray definition produces a strict angle inequality, congruence transport normalizes the two angles, and Proposition I.19 converts the angle inequality back into the desired side inequality.

The resulting Lean proof therefore reproduces the classical geometry while exposing its exact logical structure. No new axiom is introduced, and the only new general infrastructure is the left-transport theorem for strict angle comparison.