Euclid's Elements — Book I
Proposition 47

Formal Reconstruction — Technical Documentation

Jurek Różański
Military University of Technology, Warsaw, Poland
jerzy.rozanski@wat.edu.pl

CGJteamLab

The Pythagorean theorem

1 Introduction

Proposition I.47 is Euclid's theorem of Pythagoras. In the classical statement, if ABC is right-angled at A, then the square on the hypotenuse BC is equal to the sum of the squares on the legs AB and AC.

The word "equal" is crucial. The present formalization does not replace Euclid's statement by the modern numerical equation

$$a^2 + b^2 = c^2.$$

Instead, the theorem remains inside the synthetic content calculus developed for I.35–I.45. The square on BC is represented by a parallelogram term, the two leg squares are represented by two parallelogram terms, and the conclusion is

```
HilbertScissorsEquicomplementable
```

between the hypotenuse square and the formal sum of the two leg squares.

This distinction is not cosmetic. Hartshorne explicitly presents I.47 as an equal-content theorem, and Hilbert's discussion of the Euclidean theory of polygons formulates Pythagoras as equicomplementability of the three squares. The formal theorem therefore stays close to the synthetic meaning of Book I rather than importing a numerical area function.

The current Lean source is substantially longer than the classical proof. The reason is geometric placement. Euclid reads from the diagram that the three squares are constructed outward, that the perpendicular from A to BC lands inside BC, that the cut reaches the opposite side of the square, and that the auxiliary diagonal intersects the cut internally. In Lean all of these are strict order or separation statements that must be proved.

The final source is organized around four large layers:

1. construct the three outward squares;
2. construct and order the internal cut of the hypotenuse square;
3. prove the angle and noncollinearity data needed for two SAS congruences;
4. convert the congruent triangles into equal-content statements and assemble the scissors decomposition.

No proposition-local geometric axiom remains in the file, and the complete project build is clean.

2 The Classical Configuration

Let ABC be a nondegenerate triangle with a right angle at A. The Lean labelling is fixed as follows:

$$\begin{array}{ll} BCED & \text{square on the hypotenuse } BC, \\ ABFG & \text{square on the leg } AB, \\ ACKH & \text{square on the leg } AC. \end{array}$$

The square on BC is constructed on the side of the carrier BC opposite A. Likewise the square on AB is constructed on the side opposite C, and the square on AC on the side opposite B. These are not merely visual conventions. The formal theorem `i47_outward_squares` returns the corresponding `HilbertOppositeSide` data.

The standard Euclidean auxiliary line through A is parallel to DB. It meets BC at M and the opposite side DE at L. Figure 1 shows this classical configuration and deliberately stops there: the additional point N belongs to the later formal scissors construction, not to Euclid's visible diagram.

The formal reconstruction later augments this classical picture with the diagonal DC of the square on the hypotenuse and its intersection N with the cut. The strict order information required by the exact dissection is

$$B - M - C, \qquad D - L - E, \qquad D - N - C, \qquad M - N - L.$$

The two pieces of the square on BC are the parallelograms

$$LDBM \qquad \text{and} \qquad CELM.$$

They are later rotated in the scissors representation to the vertex orders needed by the final formal sum.

3 Classical Proof

Euclid's classical proof has a short dependency graph. In the usual labelling, the essential cited propositions are I.4, I.14, I.31, I.41, and I.46.

First construct the three squares by I.46. Through A draw the line parallel to BD and extend it through the square on BC; this is the I.31 step. Because the relevant angles at A are right and the two leg squares are constructed externally, I.14 gives the straight-line extensions required to identify the side parallels of the leg squares with the opposite legs of the triangle.



Figure 1: The classical configuration of I.47 in the Lean vertex order. The three squares are constructed outward from the right triangle ABC, and the auxiliary line through A is parallel to BD, meeting BC at M and DE at L. No formal crossing point N is shown at this stage.

At B, Euclid compares the triangles

$$\triangle BAD \qquad \text{and} \qquad \triangle BFC.$$

The square data give

$$BA \cong BF, \qquad BD \cong BC.$$

The angle step gives

$$\angle ABD \cong \angle FBC.$$

Hence I.4 gives

$$\triangle BAD \cong \triangle BFC.$$

Now I.41 is applied twice. The rectangle $LDBM$ is double a triangle on the base BD between the parallels BD and AL, while the square $ABFG$ is double a congruent triangle on the base BF. Consequently

$$LDBM \approx ABFG,$$

where $\approx$ denotes equal content in the present documentation.

The symmetric argument at C gives

$$MCEL \approx ACKH.$$

Finally the two rectangles make up the whole square $BCED$, and Common Notion 2, expressed in the formal scissors layer by compatibility with addition, yields

$$BCED \approx ABFG + ACKH.$$

The classical dependency graph can be summarized as

$$\begin{gathered}\text{I.46: three squares } \Downarrow \\ \text{I.14 + I.31: straight extensions and parallels } \Downarrow \\ \text{I.4: two congruent-triangle bridges } \Downarrow \\ \text{I.41: rectangle = double triangle = square } \Downarrow \\ \text{add the two equal-content relations } \Downarrow \\ \text{I.47.}\end{gathered}$$

This proof is synthetic throughout. It does not assign numbers to segment lengths or to the contents of the squares.

4 Equal Content, Not Numerical Area

Proposition I.47 belongs to the area/content block beginning at I.35. The formal statement must therefore preserve the distinction between three notions:

- geometric congruence of segments and triangles;
- scissors equality, represented by `HilbertScissorsEq`;
- equal content, represented by `HilbertScissorsEquicomplementable`.

The theorem does not assert that the square on the hypotenuse is directly `HilbertScissorsEq` to the sum of the two leg squares. The final relation is equicomplementability. This is exactly the level required by Euclid's use of equality of plane figures and by Hilbert's reconstruction of that equality through cutting and adding pieces.

A stronger scissors equality does occur internally. The theorem `i47_square_split` proves that the square on BC is actually dissected into the two parallelogram terms arising from the cut:

```
theorem i47_square_split
    [HilbertIncidence Geo]
    [HilbertEuclideanPlane Geo]
    (B C D E L M N : Geo.Point)
    (hParallelogram : IsParallelogram Geo B C E D)
    (hBMC : Geo.Between B M C)
    (hDLE : Geo.Between D L E)
    (hDNC : Geo.Between D N C)
    (hMNL : Geo.Between M N L) :
    HilbertScissorsEq Geo
      (hilbertParallelogramTerm Geo B C E D)
      (hilbertParallelogramTerm Geo M L D B +
       hilbertParallelogramTerm Geo C E L M)
```

Thus the final argument has a clean separation:

$$\begin{gathered}\text{exact dissection of the hypotenuse square } \Downarrow \\ \text{equal-content comparison of each rectangle with one leg square } \Downarrow \\ \text{addition and transport } \Downarrow \\ \text{Pythagorean equal-content conclusion.}\end{gathered}$$

No numerical area function is introduced anywhere in this chain.

5 Proof Architecture of the Reconstruction

The Lean proof is best read as a sequence of geometric packages rather than as a line-by-line translation of Euclid's text.

5.1 Step 1: construct the three outward squares

The first substantial theorem is

```
i47_outward_squares
```

It chooses the three side carriers and constructs squares on BC, AB, and AC with explicit side control. In ASCII notation, its essential output is

```
exists lBC lAB lAC : Geo.Line,
exists D E F G H K : Geo.Point,
  IsSquare Geo B C E D /\
  IsSquare Geo A B F G /\
  IsSquare Geo A C K H /\
  HilbertOppositeSide Geo D A lBC /\
  HilbertOppositeSide Geo E A lBC /\
  HilbertOppositeSide Geo G C lAB /\
  HilbertOppositeSide Geo F C lAB /\
  HilbertOppositeSide Geo H B lAC /\
  HilbertOppositeSide Geo K B lAC
```

This is stronger positional information than the public I.46 theorem alone returns. The local reusable constructor

```
hilbert_square_exists_opposite_side
```

rebuilds the I.46 construction with a prescribed choice of side. The key lower-level helper

```
hilbert_erect_equal_perpendicular_opposite_side
```

constructs the equal perpendicular side on the chosen side of the carrier.

These helpers were introduced for I.47, but their mathematical content is not specific to the Pythagorean theorem. They are plausible candidates for later promotion if another proposition requires side-controlled square construction.

5.2 Step 2: recover the straight extensions at the right-angle vertex

The outward placement of the two leg squares forces their inner sides to be opposite right-angle rays at A. The local theorem

```
i47_aux_opposite_right_rays_straight
```

packages the general fact that two right-angle rays based at the same point, on opposite sides of the supporting line, are opposite rays.

Applied to the two leg squares it yields

$$G - A - C, \qquad H - A - B.$$

These strict betweenness relations are then converted into the parallels

$$AC \parallel BF, \qquad AB \parallel CK.$$

The corresponding helpers are

```
i47_aux_square_AB_extension
i47_aux_square_AC_extension
i47_aux_square_AB_parallel_AC
i47_aux_square_AC_parallel_AB
```

Classically this is the short I.14 stage. Formally it is a plane-separation and ray-orientation argument because the straight extensions must be proved rather than read from the figure.

5.3 Step 3: prepare the two SAS congruences

The two triangle comparisons need the angle congruences

$$\angle ABD \cong \angle FBC, \qquad \angle ACE \cong \angle KCB.$$

They are established by

```
i47_aux_angle_ABD_FBC
i47_aux_angle_ACE_KCB
```

using congruence of all right angles together with angle addition and the unoriented angle representation.

Before SAS can be applied, Lean also needs four explicit noncollinearity conditions. Two are nontrivial:

```
i47_aux_noncollinear_BAD
i47_aux_noncollinear_CAE
```

They rule out degenerate placements by showing that a nondegenerate triangle cannot have two right angles. That contradiction is packaged in

```
i47_aux_two_right_angles_impossible
```

and ultimately uses I.17.

The remaining two noncollinearities follow from the already established strict parallels and are bundled by

```
i47_aux_parallel_noncollinearities
```

This layer produces exactly the geometric data consumed by the final two SAS calls. It does not yet use scissors content.

5.4 Step 4: construct the internal cut of the hypotenuse square

This is the longest new geometric component of I.47. Its final package is `i47_cut_core`.

5.4.1 The perpendicular foot

From A drop a perpendicular to the carrier BC. The helper

```
i47_aux_perpendicular_foot_on_BC
```

returns a foot M on the carrier and the corresponding right-angle data. At this point it does *not* assume that M lies between B and C.

The strict order

$$B - M - C$$

is proved separately by

```
i47_aux_perpendicular_foot_between_BC
```

The proof excludes the two exterior orders with the exterior-angle theorem I.16. The auxiliary angle lemmas

```
i47_aux_angle_ABC_less_right_BAC
i47_aux_angle_ACB_less_right_BAC
```

show that both acute angles of the right triangle are strictly smaller than the right angle at A.

This is a good example of hidden diagram data becoming a real theorem. The ordinary picture makes the altitude foot look internal; the formal proof must exclude both exterior collinear possibilities.

Figure 2 records exactly the geometric state reached after this order argument.

5.4.2 Make the cut parallel to the square side

Once $B - M - C$ is known, the perpendicular through M is shown parallel to the square side CE:

$$MA \parallel CE.$$

The helper

```
i47_aux_perpendicular_foot_parallel_CE
```



Figure 2: The first formal cut state. The perpendicular from A to the carrier BC has foot M, and the nontrivial order theorem proves $B - M - C$. The picture records the proved state; it does not serve as evidence that the foot is internal.

chooses a point between M and C, observes that the relevant angles are both right, and applies the alternate-angle criterion with the required OppositeSide data.

5.4.3 Construct the upper cut point

The point L is obtained by completing

$$C, E, M$$

to the parallelogram

$$CELM.$$

This is the role of

```
i47_aux_construct_L_parallelogram
```

Since both MA and ML are parallel to CE and pass through M, Euclidean uniqueness of parallels identifies their carriers. Thus

$$A, M, L \text{ are collinear.}$$

This is packaged by

```
i47_aux_cut_line_collinear
```

A second carrier-identification argument proves that L lies on the upper side carrier DE of the square:

```
i47_aux_L_on_DE
```

At this stage the proof has collinearity of D, L, E, but not yet the strict order $D - L - E$.

5.4.4 Recognize the left parallelogram and transfer order

Using the two known carriers and the square parallels, the proof constructs

$$LDBM$$

as a parallelogram. The theorem

```
i47_aux_left_cut_parallelogram
```

contains one delicate nondegeneracy argument: it must prove $L \neq D$ before strict parallelism can be transported. The proof uses

$$CM < CB$$

from $B - M - C$ and transports this proper-part relation through opposite-side congruences of the two parallelograms.

Now the lower-side order can be transferred to the upper side. Opposite-side congruences give

$$BM \cong DL, \qquad BC \cong DE, \qquad MC \cong LE.$$

Hilbert's three-point transfer theorem then yields

$$D - L - E.$$

This step is isolated as

```
i47_aux_upper_cut_between
```

At this point the cut has reached the opposite side and the two pieces are genuine parallelograms, but the diagonal-crossing witness N has not yet been introduced. Figure 3 keeps that intermediate state separate from the later Pasch argument.

5.4.5 Construct the crossing point N

The final order witness is the intersection of ML with the diagonal DC. The helper

```
i47_aux_diagonal_cut_intersection
```

returns

```
exists N : Geo.Point,
  Geo.Between D N C /\
  Geo.Between M N L
```

The proof applies forced Pasch to triangle CBD. The line ML enters the triangle through the open side CB at M. It cannot meet BD, because $DB \parallel ML$. Pasch therefore forces an intersection with the open side CD, giving $D - N - C$. A parallel-order theorem then gives $M - N - L$.

Figure 4 isolates precisely the new Pasch state rather than repeating the whole cut construction.

5.5 Step 5: package the complete diagram

The theorem

```
i47_diagram
```



Figure 3: The cut after constructing L. The proved orders are $B - M - C$ and $D - L - E$; the two pieces are the parallelograms $LDBM$ and $CELM$, and A, M, L lie on one carrier. The point N and the diagonal DC belong to the next stage.

is not an axiom. It is now a theorem assembling the three outward squares, the complete cut, the two leg-square parallels, the two angle congruences, and the four noncollinearity conditions.

Its role is architectural. The public theorem can destructure one geometric package and then perform the content argument without interleaving it with further plane-separation proofs.

5.6 Step 6: the two SAS bridges

The square-side congruences provide

$$BA \cong BF, \qquad BD \cong BC,$$

and the prepared angle theorem gives

$$\angle BAD \cong \angle BFC$$

in the exact orientations required by the project's SAS interface. Hence

$$\triangle BAD \cong \triangle BFC.$$

On the other side,

$$CA \cong CK, \qquad CE \cong CB, \qquad \angle CAE \cong \angle CKB$$

yield

$$\triangle CAE \cong \triangle CKB.$$



Figure 4: The final geometric step inside `i47_cut_core`. In triangle CBD, the line ML enters through CB at M and cannot meet BD because $DB \parallel ML$. Forced Pasch therefore produces N on CD, with $D - N - C$; the parallel-order argument then gives $M - N - L$.

Congruent triangles are immediately converted to `HilbertScissorsEq` by `scissors_congruent`. No area comparison is needed at the SAS stage.

Figure 5 shows the two symmetric bridges used in the content proof.

5.7 Step 7: use I.41 to pass from triangles to quadrilaterals

For the left side, I.41 is applied to the parallelogram $LDBM$ with the point A on the parallel cut carrier. It proves that the parallelogram has the content of two copies of triangle ADB.

A second application of I.41 to the square $ABFG$ with the point C proves that the square has the content of two copies of triangle CBF. Since the corresponding triangles are congruent, addition and transitivity yield

$$LDBM \approx ABFG.$$

The same argument gives

$$MCEL \approx ACKH.$$

This is the exact formal counterpart of Euclid's use of I.41. The Lean proof uses no cancellation and no converse-content theorem.

5.8 Step 8: dissect the hypotenuse square and add the two relations

The theorem `i47_square_split` cuts the parallelogram term for $BCED$ along the diagonal DC, splits the two resulting triangles at M and L, and then re-triangulates the crossing quadrilateral $DMCL$ using the intersection point N.

The result is the exact scissors equality

$$BCED = (MLDB) + (CELM)$$

at the level of `HilbertScissorsEq`.



Figure 5: The two symmetric content chains. In panel (a), the dashed triangles BAD and BFC are congruent and connect the square $ABFG$ to the cut parallelogram $LDBM$ via I.41. Panel (b) shows the corresponding triangles CAE and CKB, connecting $ACKH$ to $MCEL$.

Figure 6 displays the geometric exchange performed inside `i47_square_split`: first triangulate along DC, split at M and L, and then retriangulate the crossing quadrilateral $DMCL$ along ML before regrouping the pieces into the two cut parallelograms.



Figure 6: The exact dissection used by `i47_square_split`. Panel (a) uses the diagonal DC of the hypotenuse square. Panel (b) splits the two diagonal triangles at M and L; the central quadrilateral has crossing diagonals DC and ML meeting at N. Panel (c) exchanges the central triangulation to ML and regroups the four pieces as the parallelograms $MLDB$ and $CELM$.

The local algebraic helper

```
i47_aux_equicomplementable_add
```

combines the two rectangle-to-square equicomplementabilities. The final transport is performed by

```
equicomplementable_transport
```

which carries that content equality across the exact dissection of the hypotenuse square.

This closes the theorem without subtraction, positivity, numerical area, or a new content axiom.

6 Lean Formalization

6.1 Imports

The current file imports exactly

```
import CGJteamLab.Proposition12
import CGJteamLab.Proposition14
import CGJteamLab.Proposition17
import CGJteamLab.HilbertRightAngle
import CGJteamLab.Proposition41
import CGJteamLab.Proposition46
```

These imports already show that the formal dependency graph is not identical to Euclid's proposition list. I.12 supplies the perpendicular-from-an-external point interface used in the cut construction; I.17 supplies the two-right-angle contradiction; I.41 and I.46 are the two large direct Euclidean dependencies.

6.2 The cut-core theorem

In ASCII notation, the central geometric theorem has the form

```
theorem i47_cut_core
    [HilbertIncidence Geo]
    [HilbertEuclideanPlane Geo]
    (A B C D E : Geo.Point)
    (base : Geo.Line)
    (hABC : Not (Collinear Geo A B C))
    (hRight : HilbertRightAngle Geo B A C)
    (hBbase : HilbertIncidence.OnLine B base)
    (hCbase : HilbertIncidence.OnLine C base)
    (hSquare : IsSquare Geo B C E D)
    (hOppDA : HilbertOppositeSide Geo D A base)
    (hOppEA : HilbertOppositeSide Geo E A base) :
    exists L M N : Geo.Point,
      Geo.Between B M C /\
      Geo.Between D L E /\
      Geo.Between D N C /\
      Geo.Between M N L /\
      IsParallelogram Geo L D B M /\
      IsParallelogram Geo C E L M /\
      Geo.Parallel L A D B /\
      Geo.Parallel M A C E
```

This theorem is the main formal replacement for the positional information that Euclid leaves in the diagram.

6.3 The full diagram theorem

The theorem `i47_diagram` assembles the complete configuration. Its output includes:

- the three `IsSquare` objects;
- the four strict betweenness relations $B - M - C$, $D - L - E$, $D - N - C$, $M - N - L$;
- both cut parallelograms;
- $LA \parallel DB$ and $MA \parallel CE$;
- $AC \parallel BF$ and $AB \parallel CK$;
- the two angle congruences needed by SAS;
- the four noncollinearity facts needed by the angle and SAS interfaces.

The theorem deliberately returns more than the public statement. It is an internal geometry package from which the content proof can proceed without reconstructing positions.

6.4 The public proposition

The public theorem is

```
theorem euclid_proposition_47
    [HilbertIncidence Geo]
    [HilbertEuclideanPlane Geo]
    (A B C : Geo.Point)
    (hABC : Not (Collinear Geo A B C))
    (hRight : HilbertRightAngle Geo B A C) :
    exists D E F G H K : Geo.Point,
      IsSquare Geo B C E D /\
      IsSquare Geo A B F G /\
      IsSquare Geo A C K H /\
      HilbertScissorsEquicomplementable Geo
        (hilbertParallelogramTerm Geo B C E D)
        (hilbertParallelogramTerm Geo A B F G +
         hilbertParallelogramTerm Geo A C K H)
```

The conclusion is therefore exactly the synthetic Pythagorean statement in the current project language: existence of the three squares plus an equicomplementability relation between the hypotenuse square and the formal sum of the leg squares.

7 Relationship with Earlier Propositions

7.1 Proposition I.4

The two central triangle comparisons are SAS. In Euclid they are explicit applications of I.4; in the Lean reconstruction they are calls to the established Hilbert SAS interface. The substantial local work lies in producing the correctly oriented angle congruences and noncollinearity data before SAS is called.

7.2 Proposition I.12

The internal cut begins by dropping a perpendicular from A to the carrier BC. The project imports I.12 because the general theorem

```
hilbert_perpendicular_from_point_exists
```

is the existence mechanism used by `i47_aux_perpendicular_foot_on_BC`.

This differs from Euclid's visible I.47 dependency list. The historical proof simply draws the line through A parallel to BD. The formal reconstruction finds the same carrier by first constructing a perpendicular to BC and then using right-angle and parallel uniqueness.

7.3 Proposition I.14

I.14 is the classical straight-line step at A. The formal proof recovers its geometric effect through the stronger local theorem about opposite right-angle rays. The outward placement of the leg squares supplies the required side information, and the result is the strict orders $G - A - C$ and $H - A - B$.

7.4 Propositions I.16 and I.17

These propositions enter the hidden-order proof.

I.16 excludes the two exterior locations of the perpendicular foot M on the carrier BC. The proof makes both acute angles of the right triangle strictly smaller than the right angle at A and derives a contradiction in each exterior case.

I.17 supplies the contradiction used when a degenerate collinear placement would force a second right angle in triangle ABC. It is therefore part of the proof that the SAS triangles BAD and CAE are genuine triangles.

7.5 Proposition I.31 and Euclidean parallel uniqueness

Euclid draws the cut through A parallel to BD. In the reconstruction the existence and uniqueness of parallel carriers appear through the developed Hilbert Euclidean interface. Carrier equality is used several times:

- to identify AM with ML;
- to identify EL with the upper square carrier ED;
- to transport the final cut parallelism to the exact endpoints required by I.41.

This is one of the main places where Group IV is visible in the formal proof.

7.6 Proposition I.41

I.41 is the main content theorem consumed by I.47. It is called four times:

- rectangle $LDBM$ versus two copies of triangle ADB;
- square $ABFG$ versus two copies of triangle CBF;
- rectangle $MCEL$ versus two copies of triangle ACE;
- square $ACKH$ versus two copies of triangle BCK.

The two SAS congruences identify the doubled triangle terms, so each rectangle is equicomplementable with the corresponding leg square.

7.7 Proposition I.46

I.46 supplies the square predicate and the square-construction geometry. I.47 needs a stronger positional version: the square must be placed on a specified side of its supporting line. The helper `hilbert_square_exists_opposite_side` adapts the I.46 construction to that requirement.

Thus I.47 does not weaken or bypass I.46. It refines its existence result by adding the side choice that Euclid's diagram assumes.

8 Hilbert and Book Zero Components Used

8.1 Incidence, carriers, and strict parallelism

The reconstruction repeatedly moves between Hilbert incidence lines and the extensional `Geo.PointLine` representation used by strict parallelism. Carrier uniqueness is central to the cut construction. Two distinct carriers through the same point cannot both be parallel to a third carrier; this is used to identify AM with ML and EL with ED.

The helper family

```
ParallelSymmetry
ParallelSwapFirstLine
ParallelSwapSecondLine
ParallelCollinearLeft
collinear_parallel_trans
hilbert_parallel_transitive_distinct
```

normalizes endpoint order after those carrier identifications.

8.2 Betweenness and segment order

The formal proof uses strict betweenness in a way that the classical text does not display. The central orders are

$$B - M - C, \quad D - L - E, \quad D - N - C, \quad M - N - L.$$

The proof of $B - M - C$ uses betweenness trichotomy and angle-order contradictions. The proof of $D - L - E$ uses congruence transport across the two parallelograms and Hilbert's three-point order-transfer theorem. Segment proper-part comparison is also used to prove $L \neq D$ before strict parallel transport can be applied.

8.3 SameSide and OppositeSide

The three outward squares are selected by explicit `HilbertOppositeSide` conditions. These side data control which of the two possible perpendicular rays is used and prevent the leg squares from falling into the triangle.

OppositeSide is also required by the alternate-angle criterion used to prove $MA \parallel CE$. Thus side-of-line data are not merely construction annotations; they enter subsequent theorems as hypotheses.

8.4 Right angles and angle order

The proof uses `HilbertRightAngle` throughout. No numerical angle measure appears. Rightness is transported through angle congruence, and all right angles are compared synthetically.

Strict angle comparison enters only to prove that the perpendicular foot lies inside BC and to rule out degenerate two-right-angle triangles. The relevant infrastructure includes

```
HilbertAngleLess
hilbert_angleLess_trans
hilbert_angleLess_irrefl
hilbert_all_right_angles_congruent
hilbert_right_angle_transport
```

8.5 Pasch

The point N is not postulated as the visually obvious intersection of DC and ML. Forced Pasch constructs it as an interior intersection of the side CD of triangle CBD. Parallelism $DB \parallel ML$ excludes the alternative exit through BD.

This is one of the clearest places where formal plane order replaces an unspoken diagrammatic fact.

8.6 Scissors operations

The content layer uses the established operations

```
HilbertScissorsEq.split
HilbertScissorsEq.add
HilbertScissorsEq.trans
scissors_congruent
equicomplementable_of_scissorsEq
equicomplementable_add
equicomplementable_symm
equicomplementable_trans
equicomplementable_transport
crossing_quadrilateral_two_triangulations
```

The local helper `i47_aux_equicomplementable_add` is a packaging lemma for componentwise addition of two equicomplementability relations. It does not introduce a new content principle.

9 Formalization Notes

9.1 Geometry

Geometry is the largest layer of Proposition I.47. The formal proof must establish all of the following data explicitly:

- each square lies on the outward side of its supporting line;
- $G - A - C$ and $H - A - B$ are strict straight extensions;
- the two leg-square sides are parallel to the opposite triangle legs;
- the two SAS angles have the exact argument order required by Lean;
- all four SAS triangle orientations are noncollinear;
- the perpendicular foot M lies in the open segment BC;
- A, M, L lie on one carrier;
- D, L, E lie on the upper square carrier;
- $LDBM$ and $CELM$ are parallelograms;
- the upper order is $D - L - E$, not one of the two exterior orders;
- DC and ML meet at an interior point N with both strict betweenness relations.

None of these is inferred from coordinates. The proof remains synthetic.

9.2 Content

The content direction is forward only. Geometry and congruence produce content equalities; the proof never starts from equal content and tries to recover geometry. Consequently I.47 does not need the proper-part/positivity machinery characteristic of the converse propositions I.39–I.40.

There is also no subtraction. The final operation is addition of two rectangle-to-square equicomplementabilities after an exact scissors split of the hypotenuse square.

9.3 Representation

Three representation choices matter.

First, a square is stored as `IsSquare`, which includes an `IsParallelogram` component. I.47 uses that component repeatedly for opposite-side parallels and congruences.

Second, a triangle in the scissors layer is a term insensitive to vertex permutation. The final proof therefore contains several explicit triangle term rotations and swaps. These are representation normalizations, not new geometry.

Third, the cut parallelograms occur in different vertex orders in the geometric and scissors stages. Theorems such as `parallelogram_term_rotateOne` convert between those representations without changing content.

9.4 Axiomatic status

Proposition I.47 is Euclidean in the current project API:

```
[HilbertIncidence Geo]
[HilbertEuclideanPlane Geo]
```

Hilbert Group IV enters through uniqueness and transitivity of parallels and through the direction from parallel lines to the required angle relations. The long cut-core proof uses this Euclidean parallel infrastructure several times to identify carriers.

The order arguments, Pasch, segment congruence, angle comparison, and SAS belong to the incidence/order/congruence layers. The side-controlled perpendicular construction itself does not require numerical coordinates or an area function.

No new geometric axiom is introduced. No new content axiom is introduced. No continuity axiom is used by the final I.47 source. No Archimedean axiom is used. There is no `sorry` or `admit`.

10 Source Audit

The classical and formal sources agree on the mathematical core but organize it differently.

Euclid and the standard Book I dependency tables give the compact historical DAG

$$\mathrm{I.4}, \mathrm{rI.14}, \mathrm{rI.31}, \mathrm{rI.41}, \mathrm{rI.46} \longrightarrow \mathrm{I.47}.$$

The classical diagram suppresses most of the order and side-of-line work.

Hartshorne is particularly important for the content semantics. He treats I.35–I.48 as Euclid's theory of area in the synthetic sense of equal content. For I.47 he describes the same chain used by the formal proof: a leg square is double one triangle, that triangle is congruent to a triangle adjacent to the hypotenuse square, and I.37/I.41-style equal-content transport identifies the corresponding rectangle. Repeating the argument on the other side and adding gives the result. Hartshorne explicitly distinguishes this from the later numerical formula obtained after coordinates and an area function are introduced.

Hilbert gives an even sharper foundational control. In his discussion of the Euclidean theory of polygons he states Pythagoras as equicomplementability of the two leg squares with the hypotenuse square. This supports the project's choice of `HilbertScissorsEquicomplementable` as the public content relation rather than a real-valued area equality.

Borsuk–Szmielew and Greenberg are useful mainly as control sources for the synthetic geometry underneath the construction: perpendiculars, parallel carriers, parallelograms, congruence, and plane separation. The present proof does not import a coordinate or field model from those sources.

The Beeson proof corpus is again most useful as a warning about hidden configuration data. The formal proof must name the side choice of each square, the internal foot order, the carrier identifications, and the Pasch intersection. These are exactly the kinds of conditions that disappear in a short diagrammatic proof.

The project Book I documentation provides the immediate local context. I.41 already supplies the double-triangle content theorem, and I.46 already supplies the square predicate and construction.

I.47 is therefore the point where the construction geometry of I.46 and the forward content machinery of I.41 are combined in a single theorem.

For the required three-way audit:

- **Geometry**: substantial. Outward square placement, strict betweenness, carrier uniqueness, perpendiculars, parallels, Pasch, angle congruence, and SAS all occur.
- **Content**: forward and additive. Four uses of I.41, two congruent-triangle bridges, one exact scissors split, and final equicomplementability addition.
- **Representation**: nontrivial but separate from the mathematics. Square predicates carry parallelogram structure; triangle terms are permutation-invariant; parallelogram terms are rotated to the vertex order required by the dissection.

11 Dependency Summary

The formal dependency chain is

Hilbert incidence + order + congruence + Group IV $\Downarrow$
right angles / perpendiculars / SameSide-OppositeSide / Pasch $\Downarrow$
parallel carrier uniqueness + parallelogram infrastructure $\Downarrow$
I.12 + I.14 + I.16/I.17 infrastructure + I.41 + I.46 $\Downarrow$
`hilbert_square_exists_opposite_side` $\Downarrow$
`i47_outward_squares` $\Downarrow$
leg-square extension / parallel / angle / noncollinearity helpers $\Downarrow$
`i47_aux_perpendicular_foot_between_BC` $\Downarrow$
`i47_aux_perpendicular_foot_parallel_CE` $\Downarrow$
`i47_aux_construct_L_parallelogram` $\Downarrow$
`i47_aux_cut_line_collinear` $\Downarrow$
`i47_aux_L_on_DE` $\Downarrow$
`i47_aux_left_cut_parallelogram` $\Downarrow$
`i47_aux_upper_cut_between` $\Downarrow$
`i47_aux_diagonal_cut_intersection` $\Downarrow$
`i47_cut_core` $\Downarrow$
`i47_diagram` $\Downarrow$
two SAS congruences + four I.41 applications $\Downarrow$
`i47_square_split` + `i47_aux_equicomplementable_add` $\Downarrow$
`euclid_proposition_47`.

The contrast with the classical DAG is therefore

Euclid: I.46 + I.14 + I.31 + I.4 + I.41 $\Downarrow$
Pythagorean dissection,

while the Lean proof expands the hidden placement and order obligations into a separate geometric construction layer before entering the content argument.

New geometric axiom:	no
New content axiom:	no
New proposition-local helper:	yes
New reusable helper:	yes, side-controlled square construction
New Book Zero theorem:	no
New Interface theorem:	no in the final cleanup
Numerical area semantics:	no
Exact scissors dissection used:	yes
Equicomplementability used:	yes
Continuity used:	no
Archimedean axiom used:	no
Hilbert Group IV used:	yes
Public theorem compiles:	yes
Full `lake build`:	clean
Active local `axiom`:	no
`sorry`/`admit` in `Proposition47.lean`:	no

Diagrammatic audit. The final figure set deliberately separates the classical and formal roles. Figure 1 is the classical Euclidean configuration and therefore omits the formal crossing point N. Figures 2, 3, and 4 record three successive formal geometric states of `i47_cut_core`: internal perpendicular foot, completed upper cut, and Pasch crossing. Figure 5 records the two symmetric SAS/I.41 bridges, while Figure 6 records the genuinely geometric retriangulation inside the exact scissors split. No separate figures are introduced for triangle-term permutations, parallelogram-term rotations, or associativity/commutativity bookkeeping, since those operations do not change the geometric configuration.

12 Conclusion

Proposition I.47 is the point where several strands of the Book I reconstruction meet.

The classical proof is famously short because its diagram carries a large amount of geometric information implicitly. The formal proof exposes that information: the three squares must be placed on specified sides, the perpendicular foot must be proved internal, the upper cut point must be placed between the correct endpoints, and the diagonal must be shown to cross the cut internally. The central theorem `i47_cut_core` is the compact package of this hidden geometry.

Once the geometry is complete, the content proof is comparatively small and close to Euclid. Two SAS congruences connect the leg-square triangles with the triangles adjacent to the hypotenuse square. Four applications of I.41 turn those triangle congruences into two rectangle-to-square equal-content relations. The theorem `i47_square_split` gives an exact dissection of the hypotenuse square into those rectangles, and addition completes the argument.

The final theorem therefore preserves the genuinely synthetic form of Pythagoras:

$$\text{square on } BC \quad \approx \quad \text{square on } AB + \text{square on } AC,$$

where $\approx$ is equicomplementability, not numerical area equality.

The resulting source contains no local axiom and no unfinished proof. It is also prepared for Proposition I.48, which can now consume I.47 as an ordinary compiled theorem rather than depending on provisional diagram data.