Euclid's Elements — Book XI
Proposition 2

Formal Reconstruction — Technical Documentation

Jurek Różański
Military University of Technology, Warsaw, Poland
jerzy.rozanski@wat.edu.pl

CGJteamLab

If two straight lines cut one another, then they lie in one plane; and every triangle lies in one plane.

1 Introduction

Proposition XI.2 contains two closely related incidence statements:

1. two distinct straight lines which intersect lie in one plane;
2. every nondegenerate triangle lies in one plane.

The proposition is still entirely incidence-theoretic. No order, congruence, perpendicularity, parallelism, or continuity is needed.

Its importance is structural. Euclid wants to begin reasoning with configurations of intersecting lines and triangles in space, but such reasoning first requires an existence principle for planes. The present Hilbert reconstruction makes that existence explicit.

The production file is

```
CGJteamLab/Proposition11_2.lean
```

and the decisive reusable theorem is

```
hilbert_plane_through_two_intersecting_lines
```

from

```
CGJteamLab/Hilbert3DInterface.lean
```

2 Euclid's Statement

Euclid states:

> If two straight lines cut one another, then they lie in one plane; and every triangle lies in one plane.

Thus XI.2 has two conclusions, not merely one. In the formal development they are exposed as two public theorems:

```
euclid_proposition_11_2
euclid_proposition_11_2_triangle
```

The first handles intersecting lines. The second isolates the triangle clause.

3 Classical Configuration

Let the two straight lines AB and CD intersect at E.

Euclid then selects points on the rays EC and EB and considers the triangle ECB. The intended plane containing this triangle becomes the plane in which the two intersecting carriers AB and CD must lie.



Figure 1: The incidence configuration of Proposition XI.2. The two distinct carriers AB and CD meet at E. The triangle ECB determines the plane π; once EC and EB lie in that plane, the full carriers CD and AB lie there as well.

The essential data are therefore

$$E \in AB, \qquad E \in CD, \qquad AB \neq CD.$$

For the triangle clause, the corresponding data are simply three noncollinear points.

4 Euclid's Classical Proof

Euclid first considers the triangle ECB. He argues that no part of this triangle can lie in one plane while another part lies in another, because that would force part of one of its straight sides to lie in one plane and part in another, contradicting XI.1.

Once the triangle ECB is taken to lie in one plane, its two sides EC and EB lie in that plane. Since EC is part of the carrier CD and EB is part of the carrier AB, XI.1 is used again to promote the side segments to the whole straight lines. Hence both intersecting lines lie in one plane.

The second conclusion, that every triangle lies in one plane, is treated by the same idea.

The classical local dependency is therefore

$$\begin{gathered} AB \cap CD = \{E\} \\ \Downarrow \\ \text{consider triangle } ECB \\ \Downarrow \\ \text{XI.1} \\ \Downarrow \\ ECB \text{ lies in one plane} \\ \Downarrow \\ EC, EB \text{ lie in that plane} \\ \Downarrow \\ \text{XI.1} \\ \Downarrow \\ AB, CD \text{ lie in that plane.} \end{gathered}$$

5 The Missing Plane-Existence Step

The classical argument contains a foundational difficulty. It speaks of "the plane of reference" and reasons about parts of the triangle lying in that plane, but no plane containing the relevant spatial configuration has yet been constructed.

This is exactly the sort of obligation that a modern incidence axiomatization must make explicit.

A natural repair is the Hilbert incidence principle that three noncollinear points determine a plane. Once such a plane exists, XI.1-like line closure can be applied rigorously.

The formal reconstruction therefore does not imitate the circular "reference plane" language. It explicitly produces the required plane from incidence data.

6 What Is Genuinely Three-Dimensional Here?

XI.2 is the first Book XI proposition whose conclusion is an existential statement about a plane determined by spatial data.

For the intersecting-line clause the input consists of two ambient lines

$$l, m$$

and a common point

$$P \in l \cap m, \qquad l \neq m.$$

The conclusion is the existence of an ambient plane

$$\pi$$

such that

$$l \subset \pi, \qquad m \subset \pi.$$

For the triangle clause the input is a noncollinear triple

$$A, B, C,$$

and the conclusion is a plane containing all three points.

No induced planar theorem is needed after the plane has been obtained. Thus XI.2 is genuinely spatial, but purely at the incidence level.

7 Formal Statement: Intersecting Lines

The first production theorem is:

```
theorem euclid_proposition_11_2
    [H : HilbertIncidence Geo]
    [HilbertPlaneIncidence Geo]
    [S : HilbertSpacePrimitive Geo]
    [HSI : HilbertSpaceIncidence Geo]
    (l m : Geo.Line)
    (P : Geo.Point)
    (hlm : Ne l m)
    (hPl : H.OnLine P l)
    (hPm : H.OnLine P m) :
    exists pi : S.Plane,
      HilbertLineInPlane Geo l pi /\
      HilbertLineInPlane Geo m pi := by

  have h :=
    hilbert_plane_through_two_intersecting_lines
      (Geo := Geo)
      l m hlm P hPl hPm

  cases h with
  | intro pi hData =>
      exact
        Exists.intro pi
          (And.intro
            hData.1
            hData.2.1)
```

The theorem intentionally exposes only the existence conclusion required by Euclid.

The reusable Hilbert theorem is stronger: it also proves uniqueness of the plane.

8 The Stronger Hilbert Theorem

The interface theorem has the form

```
theorem hilbert_plane_through_two_intersecting_lines
    [H : HilbertIncidence Geo]
    [HilbertPlaneIncidence Geo]
    [S : HilbertSpacePrimitive Geo]
    [HilbertSpaceIncidence Geo]
    (l m : Geo.Line)
    (hlm : Ne l m)
    (P : Geo.Point)
    (hPl : H.OnLine P l)
    (hPm : H.OnLine P m) :
    exists pi : S.Plane,
      HilbertLineInPlane Geo l pi /\
      HilbertLineInPlane Geo m pi /\
      forall rho : S.Plane,
        HilbertLineInPlane Geo l rho ->
        HilbertLineInPlane Geo m rho ->
        rho = pi
```

Thus the Hilbert layer proves not only

$$\exists \pi, \quad l, m \subset \pi,$$

but also

$$\forall \rho, \quad l, m \subset \rho \Longrightarrow \rho = \pi.$$

The Euclid theorem discards this uniqueness component because it is not part of XI.2's stated conclusion.

9 How the Stronger Theorem Is Proved

The reusable theorem is a genuine derived incidence result.

Starting from the common point P, it chooses a second point A on l and a second point B on m, with

$$A \neq P, \qquad B \neq P.$$

Because $l \neq m$, the three points A, P, B are noncollinear. Hilbert spatial incidence then supplies a plane through them.

The two line-containment conclusions follow from the XI.1-type incidence principle:

$$A, P \in l \cap \pi \Longrightarrow l \subset \pi,$$

$$B, P \in m \cap \pi \Longrightarrow m \subset \pi.$$

Finally, if another plane ρ contained both l and m, then it would contain the same noncollinear triple A, P, B. Plane uniqueness therefore gives

$$\rho = \pi.$$

Schematically,

$$\begin{array}{c}
l \neq m, \quad P \in l \cap m \\
\Downarrow \\
A \in l,\ A \neq P, \qquad B \in m,\ B \neq P \\
\Downarrow \\
A, P, B \text{ noncollinear} \\
\Downarrow \\
\text{Hilbert I.4} \\
\Downarrow \\
\exists \pi:\ A, P, B \in \pi \\
\Downarrow \\
\text{Hilbert I.6} \\
\Downarrow \\
l, m \subset \pi \\
\Downarrow \\
\text{Hilbert I.5} \\
\Downarrow \\
\pi \text{ is unique.}
\end{array}$$

This is the real formal incidence content behind the short public wrapper.

10 Formal Statement: Triangle Clause

The second public theorem is:

```
theorem euclid_proposition_11_2_triangle
    [H : HilbertIncidence Geo]
    [S : HilbertSpacePrimitive Geo]
    [HSI : HilbertSpaceIncidence Geo]
    (A B C : Geo.Point)
    (hABC : Not (PrimCollinear Geo A B C)) :
    exists pi : S.Plane,
      S.OnPlane A pi /\
      S.OnPlane B pi /\
      S.OnPlane C pi := by

  exact
```

```
    HilbertSpaceIncidence.plane_through
      (Geo := Geo)
      A B C hABC
```

This is precisely the existence part needed to say that a nondegenerate triangle lies in a plane.

The theorem returns the three vertex-incidence facts. If explicit side carriers are later introduced, Hilbert I.6 promotes each pair of vertex incidences to whole-line containment.

11 Why the Triangle Theorem Is Minimal

It would be possible to formulate a larger theorem returning explicit side lines AB, BC, and CA together with proofs that all three lie in the same plane.

The current theorem deliberately avoids that extra representation layer. At the incidence level a nondegenerate triangle is already determined by three noncollinear vertices. The existence of their plane is the exact content of the second clause of XI.2.

Whenever side carriers are required, they can be constructed separately by the ordinary two-point line theorem and placed in the same plane using `line_in_plane`.

12 Spatial Architecture

The intersecting-line theorem uses only:

```
HilbertIncidence Geo
HilbertPlaneIncidence Geo
HilbertSpacePrimitive Geo
HilbertSpaceIncidence Geo
```

The triangle theorem uses an even smaller signature:

```
HilbertIncidence Geo
HilbertSpacePrimitive Geo
HilbertSpaceIncidence Geo
```

There is no use of

```
HilbertSpaceOrder
HilbertSpaceCongruence
HilbertSpaceEuclidean
PlaneGeo
```

Thus XI.2 belongs completely to Hilbert's spatial Group I incidence layer.

13 Hidden Formal Data

Several facts suppressed by the classical diagram must be explicit in Lean.

13.1 The two carriers must be distinct

The hypothesis

```
hlm : Ne l m
```

is essential. If $l = m$, infinitely many planes may contain that one line, so the unique-plane theorem is false.

13.2 The common point must be explicit

The intersection is represented by

```
P : Geo.Point
hPl : H.OnLine P l
hPm : H.OnLine P m
```

rather than by a primitive relation saying merely that the lines intersect.

13.3 Noncollinearity must be proved

The stronger helper does not assume three ready-made noncollinear points. It constructs one auxiliary point on each line and derives noncollinearity from the distinctness of the two carriers.

13.4 The witness plane survives

The conclusion is existential:

```
exists pi : S.Plane, ...
```

so the plane is a genuine proof object, not an implicit background surface supplied by the diagram.

14 Relationship with Proposition XI.1

Classically XI.1 is the immediate dependency of XI.2.

Euclid repeatedly reasons that if part of a side lies in a plane, then the whole straight line must lie there. That is precisely the content of XI.1.

The formal DAG is organized differently. The public theorem `euclid_proposition_11_2` does not call `euclid_proposition_11_1`. Instead, the stronger general theorem `hilbert_plane_through_two_inter` uses the underlying Hilbert I.6 field directly.

Hence:

$$\text{classical dependency: XI.1} \to \text{XI.2},$$

whereas

$$\text{formal dependency: Hilbert I.2–I.6} \to \texttt{hilbert\_plane\_through\_two\_intersecting\_lines} \to \text{XI.2}.$$

This is not a loss of historical structure. It records the fact that XI.1's mathematical content has already been moved into the reusable spatial incidence interface.

15 Relationship with Later Book XI Propositions

XI.2 is a basic coplanarity theorem and becomes part of the classical infrastructure for later Book XI arguments.

In particular, later propositions repeatedly need to know that two intersecting carriers determine a common plane before a planar theorem can be applied to them.

In the formal development the stronger interface theorem often replaces direct calls to XI.2. This is analogous to XI.1: the Euclid-numbered theorem records the correspondence with the *Elements*, while downstream Lean proofs are free to consume the stronger reusable result.

16 Axiomatic Status

The audit file contains

```
import CGJteamLab.Proposition11_2

#print axioms Geometry.euclid_proposition_11_2
#print axioms Geometry.euclid_proposition_11_2_triangle
```

Lean reports

```
'Geometry.euclid_proposition_11_2' depends on axioms:
[propext, Classical.choice, Quot.sound]

'Geometry.euclid_proposition_11_2_triangle'
does not depend on any axioms
```

The two outputs have different meanings.

The first public theorem reuses the derived interface theorem for two intersecting lines. Its proof term therefore inherits the logical dependencies appearing in that developed infrastructure:

```
propext
Classical.choice
Quot.sound
```

No project-specific geometric axiom constant is reported.

The triangle clause is a direct projection from the explicit parameter

```
[HSI : HilbertSpaceIncidence Geo]
```

and therefore has no additional global axiom dependency.

As in XI.1, the phrase

```
does not depend on any axioms
```

must not be read as saying that the theorem has no geometric assumptions. The spatial incidence structure is an explicit theorem parameter.

At this checkpoint:

- new proposition-specific geometric axiom: no;
- new spatial axiom declaration: no;
- new continuity assumption: no;
- Hilbert Group II order: not used;
- Hilbert Group III congruence: not used;
- Hilbert Group IV Euclidean parallelism: not used;
- `PlaneGeo`: not used;
- new proposition-local helper theorem: no;
- new reusable `Hilbert3DInterface` theorem: no at this proposition checkpoint; the stronger helper already existed;
- `sorry`/`admit`: no;
- public theorem compiles: yes;
- axiom audit: completed.

17 Classical DAG versus Formal DAG

The distinction is especially clear for XI.2.

Euclid.

$$\begin{gathered} AB, CD \text{ intersect at } E \\ \Downarrow \\ \text{triangle } ECB \\ \Downarrow \\ \text{XI.1} \\ \Downarrow \\ \text{triangle lies in one plane} \\ \Downarrow \\ \text{XI.1} \\ \Downarrow \\ AB, CD \text{ lie in that plane.} \end{gathered}$$

The weak point is that the plane itself is never produced from an explicit plane-existence principle.

Lean / Hilbert.

$$\begin{gathered} l \neq m, \quad P \in l \cap m \\ \Downarrow \\ \text{two auxiliary line points} \\ \Downarrow \\ \text{noncollinear triple} \\ \Downarrow \\ \text{Hilbert I.4} \\ \Downarrow \\ \text{explicit plane witness } \pi \\ \Downarrow \\ \text{Hilbert I.6} \\ \Downarrow \\ l, m \subset \pi \\ \Downarrow \\ \text{Hilbert I.5} \\ \Downarrow \\ \pi \text{ unique} \\ \Downarrow \\ \texttt{euclid\_proposition\_11\_2}. \end{gathered}$$

For the triangle clause the formal DAG is even shorter:

$$A, B, C \text{ noncollinear} \Longrightarrow \text{Hilbert I.4} \Longrightarrow \exists \pi,\ A, B, C \in \pi.$$

18 Structural Lesson

XI.2 makes explicit why solid geometry cannot begin merely by reusing planar reasoning in an unspecified ambient setting.

Before one can say that two intersecting spatial lines are a planar configuration, one must prove the existence of the plane containing them. Before one can treat a triangle as planar, one must prove the existence of the plane through its vertices.

The Hilbert incidence system supplies exactly these missing existence and uniqueness principles. Consequently XI.2 becomes a short Euclid interface over a stronger and reusable spatial theorem.

This proposition is therefore an early example of a recurring pattern in Book XI:

$$\text{diagrammatic coplanarity} \quad \longrightarrow \quad \text{explicit existential plane witness.}$$

That distinction becomes increasingly important in later propositions, where several different planes coexist and their identities or intersections can no longer be left implicit.

19 Status

Production theorem	`Geometry.euclid_proposition_11_2`
Triangle theorem	`Geometry.euclid_proposition_11_2_triangle`
Production file	`CGJteamLab/Proposition11_2.lean`
General 3D interface	`CGJteamLab/Hilbert3DInterface.lean`
Coordinates/vectors	no
PlaneGeo	no
Hilbert Group II	no
Hilbert Group III	no
Hilbert Group IV	no
Continuity	no
New XI.2-specific axiom	no
Axiom audit	completed

20 Conclusion

Proposition XI.2 is the first explicit plane-existence theorem of the Euclidean solid-geometry sequence.

Its two conclusions are

$$l \neq m, \quad P \in l \cap m \quad \Longrightarrow \quad \exists \pi,\ l, m \subset \pi,$$

and

$$A, B, C \text{ noncollinear} \quad \Longrightarrow \quad \exists \pi,\ A, B, C \in \pi.$$

The historical proof tries to obtain these conclusions through XI.1 while speaking of a reference plane that has not itself been produced. The Hilbert reconstruction resolves the issue at the incidence level: noncollinear points determine a plane, two plane points force their whole line into the plane, and three noncollinear common plane points determine that plane uniquely.

The public Euclid theorem is therefore short, but unlike XI.1 its first clause sits on top of a genuine derived spatial theorem. That stronger theorem is the correct reusable form for the remainder of the formal development.